Bevezetés

A Prolog a programozás egyik legizgalmasabb kalandja, de minden bizonnyal nem csak kalandorok számára nyújt szellemi kihívást. Ráadásul az úgynevezett ötödik generációs számítógépek tervezett programozási technikája is a Prolog nyelvbõl indul ki, így ennek elsajátítása nem túl gyorsan avuló, értékes tudás. 

Az utóbbi évtizedekben a hardware árak körülbelül négyévenként megfelezõdtek, míg az írott software ára évrõl évre emelkedik, és hovatovább messze a legnagyobb részét teszi ki egy teljes rendszer árának. A software részaránya a teljes rendszer árához viszonyítva 1970-ben 10%, 1975-ben 50%, 1985-ben pedig már több, mint 80%. 

Ez a gyors árnövekedés serkentette a programozást könnyebbé, gyorsabbá tevõ eszközök kifejlesztését, olcsóbbá téve ezzel a programfejlesztést. E célból a kutatás arra irányult, hogy a programozási munka nagyobb részét maga a gép végezze el. A PROLOG ennek a sokéves kutatómunkának az eredménye. 

Az elsõ hivatalos PROLOG verziót a marseilles-i egyetemen fejlesztették ki Alain Colmerauer vezetésével az 1970-es évek elején, õk adták a nevét is "PROgraming in LOGic". Ez sokkal eredményesebb és hatékonyabb lett, mint a legtöbb jól ismert programozási nyelv, köztük a Pascal és a BASIC. Például egy adott alkalmazás PROLOG-ban általában tízszer kevesebb programsort igényel, mint Pascal-ban. 

Ekkorra tehát már nagyon felerõsödött az az igény, hogy minél több programfejlesztési munkát bízzanak a gépre, kiváltva a szellemi munkának a kevéssé hatékony, "favágó" területeit. Olyan nyelv gondolatai körvonalazódtak, amely szabályok és tények alapján dolgozik, látszólag önállóan oldja meg a problémát. 

Ma a PROLOG talán a legfontosabb eszköze a mesterséges intelligenciát alkalmazó programozásnak és a szakértõi rendszereknek. Számos, külföldön jól ismert szakértõi rendszer íródott PROLOG-ban. Az egyre inkább felhasználó-barát és intelligens programok iránti igény pedig egy további oka a PROLOG növekvõ népszerüségének. 1983-ban Japánban hirdették meg azt az ambíciózus project-et, amely az ötödik generációs számítógép tervezését és kivitelezését tartalmazza, és amely számára a PROLOG az alap-rendszer nyelv (kiegészítve a hordozó berendezés assembly nyelvének használatával). 

A jelenlegi állapotban még nem létezik ötödik generációs gép, ezért a mai Prolog implementációk Neumann-elvü gépeken müködnek. E sajátos helyzet miatt (nincs párhuzamos feldolgozás, viszonylag lassú, soros memóriák, stb.) ezek az implementációk az eredeti elképzelésekhez viszonyítva lassúak, hiányosak, sokszor nehezen kezelhetõek. Pascalban pontosan meg kell mondani, hogy hogyan kell a számítógépnek megoldania a feladatot. 

Szemben a Pascallal, a PROLOG program a probléma leírását adja meg a számítógépnek, kellõ számú tény és szabály felsorolásával, majd kéri a rendszert, hogy keresse meg a probléma összes lehetséges megoldását. Ha tehát egyszer a PROLOG programozó leírta, hogy MIT kell kiszámítani, a PROLOG rendszer maga szervezi meg, hogy HOGYAN kell a számítást végrehajtani. E deklaratív (nem procedurális) megközelítés miatt a Pascal- és BASIC-beli jól ismert hibaforrások -mint például a túl sokszor vagy túl kevésszer végrehajtott ciklusok- eleve ki vannak zárva. Ezen túlmenõen a PROLOG kikényszeríti, hogy a programozó a problémát alaposan elemezze, azt struktúrált módon tudja leírni, ezért a PROLOG specifikációs segédeszközként is használható. 

A TURBO-PROLOG rendszer az IBM-PC és vele kompatibilis gépeken futó olyan implementáció, amely azzal tünik ki, hogy teljes a maga nemében. A rendszer része a szövegszerkesztõ, amely ASCII-file-t készít, és így helyettesíthetõ bármely szokványos szövegszerkesztõvel. A fordító képes a memóriába és a lemezre is fordítani, és meglehetõsen gyors. A lefordított program futása elég gyors, hivatásos felhasználók igényeit is képes kielégíteni. Vállalati, döntéselõkészítést segítõ programcsomagok, szakértõi rendszerek is készúltek ezzel a rendszerrel. Kiváló a program belövését segítõ TRACE funkció, a kúlönbözõ rendszer-ablakok (EDITOR, TRACE, MESSAGE, DIALOG, stb), a színek, felhasználói ablakok, a file-kezelés, stb. 

A programcsomaghoz adott felhasználói kézikönyv alapos, kezdõk is megtanulhatják a használatát, és oktató jelleggel még a Prolog nyelvü programozásba is bevezet. Mivel angol nyelvü és általában kevésbé hozzáférhetõ, indokoltnak tartom, hogy legyen olyan forrás, amelybõl a magyar felhasználó gyorsabban boldogul. 

Ez a mü speciálisan azoknak szól, akik valamely programnyelven már az alapokat elsajátították, és a fent említett kalandos útra kívánnak lépni. Kúlönösen ajánlom középiskolai számítástechnikával foglalkozó tanároknak, diákoknak, hiszen ez a korosztály már eleget tudhat a számítógéprõl, és még fogékony az újra. Ajánlom azonban, hogy ne megszokásból, kizárólag a "jól bevált" sémák szerint gondolkodjanak, hanem a tiszta logika vezéreljen a problémák specifikációjában. Ez a nyelv sokkal inkább épít a kreatív gondolkodásra, mint a hagyományos nyelvek, és rutinból programozni Prologban -egy könyv címével ellentétben- nem képzelhetõ el. 

Végül egy hasonlattal ecsetelném az elõttünk álló nehézségeket. A Prolog programozás hasonló a zeneszerzéshez. A program müködése a zenemü egyszerüségével, szépségével eltakarja az alkotás kínjait, és minél szebb a mü annál inkább teszi ezt. Ahogyan a zenében egy akkord, a Prolog programozásban a szabály alkalmazása egy pillanat müve, de amíg a megfelelõ akkord vagy szabály megfogalmazódik, a legtöbb esetben sok gyötrõdést kell átélnie az alkotónak. Ez nem elriasztás akar lenni, hanem az alkotókedv felcsigázása... No, kezdjük .... 

I. FEJEZET
AZ ELSÕ LÉPÉSEK

I.1.A Turbo-Prolog betöltése, indítása és menürendszere.

Abból indulunk ki, hogy a PC merevlemezes tárolójának egy alkönyvtárában van a teljes rendszer, amin belül persze még lehetnek további (felhasználói, oktató, stb.) alkönyvtárak. Itt a 

C:>prolog
billentyûzéssel betöltõdik és el is indul a program. Pár másodperc múlva egy menüsor jelenik meg a képernyõn, és néhány ablak. Az alsó sor pedig az aktuális legfontosabb parancsok jelei, melyek zöme mindig valamelyik F funkcióbillentyû. 
A menü a szokásos módon, a vízszintes kurzorvezérlõ billentyûkkel és az ENTER lenyomásával mûködik, amire általában egy újabb, most már lefutó menüben találjuk magunkat. Ezek közül a függõleges kurzorvezérlõkkel és az ENTER-rel választhatunk. 
A programozás elsõ (technikai) lépése a program szövegének elõállítása. Erre használjuk az EDIT funkciót. A program lefordítása a COMPILE funkcióval lehetséges. Hiba esetén kijelzi a hiba helyét és az észlelt hibát, bár ez nem garancia arra, hogy az igazi hiba is ott van. Hiba esetén -még a menüpont hatásköre alatt- kijavíthatjuk a hibát, az F10-re pedig befejezi a fordítást. 
A memóriában lévõ program futtatása a RUN opció alatt történik. Ha elõtte nem fordítottuk le a programot, vagy a legcsekélyebb mértékben megváltoztattuk, akkor a futást megelõzi a fordítási procedúra. A futási kép a legegyszerûbb esetben a DIALOG nevû ablakban történik. A programszöveg mentésére, betöltésére, átnevezésére, az aktuális drive és alkönyvtár változtatására és az operációs rendszerbe történõ ideiglenes kilépésre a FILE funkció áll rendelkezésünkre. 
A Turbo-Prolog program kezelésérõl nem ejtünk több szót, mert nem lenne hatékony. Az olvasó szánjon rá egy kis idõt, hogy a funkciókat könnyedén elõvarázsolja, és ne ez okozzon problémákat. Nem lehet elég sokszor hangsúlyozni azonban, hogy a minimális angol szaknyelvi ismeretek mennyire hasznosak a PC használatában. 
  

I.2. Logikai nyelvekre vonatkozó néhány definíció

Most néhány fontos definíció és formai szabály következik, a továbbiak pontosabb megértése érdekében. 
A logikai program alapstruktúrája a kifejezés. A kifejezés lehet konstans, változó vagy összetett kifejezés. A konstansok számok vagy elemi (atomi) jellegûek lehetnek, míg a változók egyedi, de nem specifikált egyedeket jelentenek. Az "atomok" szimbóluma bármilyen karakterlánc lehet, amely kizárja bármely más (változó vagy szám) szimbólummal való összetévesztés lehetõségét. A változók neveit a kezdõ nagybetû különbözteti meg. 
A szabály egy függvénynevet (a szabály megkülönböztetõ jelét) és egy vagy több kifejezést, mint argumentumot tartalmaz. A szabály jellemzõi tehát a neve, amely tovább nem bontható, és a kiterjedése (dimenziója), vagyis az argumentumainak a száma. A tények nulla változós szabályoknak tekinthetõk. 
Szintaktikailag a szabály formája: 

f(t1,t2,...,tn) ahol a szabály neve f, dimenziója n, és az argumentumai ti.

A logikai program szabályok véges készlete. A feltétel vagy szabály 

A :- B1,B2,...,Bk. k>0
általános alakú mondatok, ahol A és Bi célkifejezések. Az ilyen mondatok kijelentõ módú deklaratív olvasása: A tartalmazza Bi együttes teljesülését. Végrehajtás szempontjából procedurálisan ugyanez így hangzik: az A kérdésre válaszolni annyi, mint válaszolni a B1,B2,...,Bk kérdésekre. "A" a szabály feje, míg Bi a teste. 
Ha k=0, akkor a szabályt ténynek nevezzük, és A.-tal jelöljük. 
Ha k=1, akkor a szabályt iteratív szabálynak nevezzük. 
A kérdés az A1,...,An? n>0 alakú kifejezés, ahol A célkifejezések. A kérdésben lévõ változók értelme azok létezésének meghatározása. 
A P logikai program lefuttatása egy olyan eset keresése, amely az adott kérdésre a programból logikailag következik. A G célállítás következik a P programból, ha a G-nek van olyan A esete, ahol 

A:-B1,...,Bn, n>0,
A a P program egy szabályának alaphelyettesítése és Bi-k a P programból következnek. 

I.3. A Prolog-program jellemzése

A jól ismert Pascal, a BASIC és a többi hagyományos programozási nyelv procedurális: a programozónak kell lépésrõl lépésre leírnia az eljárásokat, amelyek megmondják a számítógépnek, hogy hogyan kell a problémát megoldani. Ezzel szemben a PROLOG deklaratív nyelv. Ez azt jelenti, hogy megadva a szükséges tényeket és szabályokat, képes deduktív következtetésekkel megoldani a programozott problémákat. 
A PROLOG-programozónak csak a probléma leírását (goal=cél) kell felállítania a megoldás alapszabályaival, és a PROLOG rendszer fogja a megoldáshoz vezetõ utat meghatározni. A célkifejezésrõl elõbb-utóbb kideríti a Prolog rendszer, hogy az adott körülmények között teljesül-e vagy nem, illetve milyen feltétellel teljesülhet. 
A célkifejezés vizsgálatának tehát kétféle módja van: 

· Ha a célkifejezésben lévõ szabály csak konstansokat tartalmaz, akkor a rendszernek nincs más dolga, mint addíg helyettesítgetni a megengedett tényállításokat a beépített backtrack algoritmus segítségével, amíg a szabály nem teljesül, amikoris válaszként közli, hogy TRUE. Ellenkezõ esetben, tehát ha az összes lehetõség kimerült, és még mindig nem volt sikeres a keresés, a válasz FALSE. 

· Ha a célkifejezésben van változó, akkor a rendszernek az a feladata, hogy ennek keressen legalább egy olyan értéket, amely kielégíti a célt. Bizonyos körülmények között az összes megoldást megkapjuk. 

Hogyan kapcsolódik össze procedurális és a deklaratív gondolkodásmód a logikai program építésénél? Célszerûségbõl általában mindenki procedurálisan gondolkodik, amikor programoz. Azonban deklaratívan gondolkodik, amikor a program kívánt és a megvalósított jelentésének eredményét veti össze. Ezek keverésének egy módja a logikai programozásban a program szabályait procedurálisan összeállítani, majd az eredményt mint deklaratív állítást értelmezni. Építsük fel tehát procedurális gondolkodásmóddal a programot, majd mérlegeljük, hogy a különbözõ alkalmazások során az milyen deklaratív értelmet kap. Ezt majd a feladatokon keresztül be is mutatjuk. 

I.4. A Turbo-Prolog program szerkezete

Nagy hatással volt a programok világos szerkezetének kialakításában a felülrõl lefelé való kivitelezés metódusának hangsúlyozása, amit a lépésenkénti finomítás elvével együtt használunk. Nagy vonalakban ez a metódus azt jelenti, hogy felállítjuk a fõ problémát, széttördeljük alproblémákra, majd megoldjuk ezeket a részeket. A felülrõl lefelé való programozás természetes módja a logikai programok felépítésének. A közölt programleírások is kizárólag felülrõl lefelé haladnak. Az elvek rögzítése után részletesen bemutatom a Turbo-Prolog programok konkrét összetevõit is. 
Egy Prolog program tények és szabályok alapján keresi a megoldást. Ehhez ismernie kell a szabályok szerkezetét, és a szerkezeti elemek minõségét. Az igazán teljes Prolog rendszerek ezt automatikusan elvégzik, bár így sok hiba rejtve is maradhat. Valószínûleg a takarékos memóriagazdálkodás kényszerûsége és a hibák korai kiszûrése miatt a Turbo-Prolog megköveteli az elemi objektumok és a szabályok formai deklarációját. 
Ennek két lépcsõje: 

domains 

Az elemi objektumok típusainak deklarációja. Ez a rész nem kötelezõ. Ha a szabályok leírásánál (predicates, lásd késõbb) csak a rendszer által biztosított alaptípusokkal definiálunk, akkor kihagyható. Ellenkezõ esetben -és a beszédes változónevek elve betartásakor méginkább- nem kerülhetjük meg. Az sem árt, hogy a paraméterek átadásakor ezáltal elkerülhetjük a nem kívánt adatkeveredést is. 

Az elemi objektumok nevei kisbetûkkel kezdõdnek. A név és egy egyenlõségjel után vagy 

· egy már létezõ standard változótípus nevét, vagy 

· egy konstanskifejezést kell tenni, ahogyan az az alábbi példák mutatják. 

A standard típusok a következõk: 

integer, real, char, symbol, string. 

Például 

elem = symbol 

szam = integer 

nagyszam = real 

i,j,k = integer 

rud = kozepso 

Ezek után tehát például az elem szó felhasználható a szabályok megfogalmazásánál olyan pozíciókban, ahol szimbólumokat kívánunk elhelyezni, illetve ahol szimbólumokat várunk eredményként. Az elmondottak szerint a rud olyan típus, ahol az egyetlen behelyettesíthetõ érték a kozepso szó. 

Ennek persze nem sok értelme lenne, ezért rögtön ejtsünk szót egy fontos lehetõségrõl, amelyet alternatív deklarációnak nevezhetünk. Itt olyan deklarációról van szó, ahol egy típusnév többféle típust jelenthet. Legegyszerûbben ezt konstansokkal szemléltethetjük: 

rud = jobboldali ; kozepso ; baloldali 

Most a rud típus háromféle értéket kaphat, ezeket a pontosvesszõ választja el egymástól. Az alternatív deklaráció kezelése nagy óvatosságot, gyakorlatot igényel. 

A meglévõ típusokkal újabb deklarációk építhetõk fel, mint azt az alábbi példa mutatja: 

elem = symbol 

ujelem = elem 

Mód van arra, hogy egy meglévõ típust felhasználjunk összetettebb adatszerkezet elõállítására. Ennek legegyszerûbb és elõnyben részesített módja a lista definiálása. 

Ha egy szabály egyik paramétere nem elemi objektum, hanem egy lista, akkor definiálni kell egy ilyen típust a következõ módon: 

adat = symbol* 

Az adat nevû típus ezek után olyan adatszerkezet, amelyben symbol típusú elemi objektumok vannak listaszerûen összefûzve. Az sem helytelen, ha a lista helyett a verem fogalmával azonosítjuk a Prolog-beli listát, mert közvetlenül csak a lista elsõ tagja (a verem legfelsõ eleme) kezelhetõ, illetve választható le a listáról, tehát inkább veremszerûnek képzelhetõ el ez az adatszerkezet. Jelölésmódja egyszerû: ha az elemeket a,b,c,d-vel jelöljük, akkor elfogadottak az alábbiak: 

["a","b","c","d"] 

["a"|["b","c","d"]] 

["a"|["b"|["c"|["d"]]]] 

Az üres lista jele a [], amely minden lista végére implicite odaértendõ. 
  

predicates 

A szabályok felsorolása a formalizmus rögzítésével. Ebben a szekcióban azokat a szabályokat kell bevezetnünk, amelyeket alkalmazni fogunk. Ezzel rögzítjük az alkalmazott szabályok neveit és a használható paraméterek típusait. Legalább egy ilyen szabálynak kell lennie, máskülönben nem is beszélhetünk Prolog programról. Néhány példa: 

szabaly1(symbol) 

szabaly2(integer,integer,symbol) 

masik_szabaly(real,symbol) 

uj_szabaly(adat,symbol,adat) 

A deklarációval foglalkozó szekciók után most a további program-egységek következnek. Ezek közül az elsõ a program célját van hivatva rögzíteni. 

goal 

A program addíg fut, amíg ez a cél nem teljesül, vagy meg nem állapítja, hogy sosem teljesülhet. Ez a rész is elmaradhat, de a léte vagy nemléte erõsen befolyásolhatja a program mûködését. Ha van, akkor pontosan azt teszi, amire az ezt követõ utasítások kényszerítik, és az elsõ kielégítõ eset után leáll (hacsak másra nem utasítjuk). Ha nincs, akkor a menübõl választott RUN parancsra a DIALOG ablakban megjelenik a 

GOAL=? 

kiírás, ami után az aktuális cél, másnéven a kérdés begépelése következhet. Maga a program tehát csak egy bizonyos kérdés megválaszolásáért hajlandó futni. 

A cél elérésére két fontos, a nyelvbe beépített mechanizmus mûködik. Az egyik a mintaillesztés, amely két kifejezés (az elérendõ cél és a rendelkezésre álló szabályok egyike) közötti, argumentumonkénti összehasonlításra szolgál. A másik a backtrack algoritmus, amely a megoldáshalmaz összes elemét igyekszik megtalálni. Ha a lehetséges összes (jó és rossz) megoldást egy bináris fa ágaiba és leveleibe helyezzük, akkor az említett algoritmus a fa módszeres bejárását végzi, minden megoldás után visszalépve az utolsó elágazásra ahonnan újra próbálkozik másik megoldást találni. 
  

clauses 

A Prolog program utolsó része a tények és szabályok felsorolása. A tények és szabályok tetszõleges sorrendben követhetik egymást, az azonban fontos, hogy az azonos nevû szabályok egymást követõen álljanak, az azon belüli sorrend viszont már a végrehajtás sorrendjét fogja befolyásolni. Ennek az az oka, hogy számítógépeink egyszerre csak egy dologgal tudnak foglalkozni, lévén, hogy a mikroprocesszor szekvenciális mûködésû és nem tesz lehetõvé párhuzamos feldolgozást. Ha majd olyan gépen fut a Prolog, ahol ez nem korlátoz, akkor a szabályok sorrendjére és elhelyezkedésére valószínûleg nem kell ilyen megkötést tenni. 

A szabályok formai megjelenésben lényegében kétfélék. 

Az egyik neve tényállítás. Ez konstans argumentumú feltétel nélküli szabály, a szabály nevével és argumentumlistával, a végén egy pont. A másik egy olyan kifejezés, amely egy vagy több kisebb célállítás logikai kapcsolatából áll. A részcélok argumentumlistáiban szerepelhet egy vagy több változó, és az azonos nevû változónevek egy szabályon belül mindig ugyanazt a behelyettesítést jelentik. A szabály teljesülése feltételektõl függ. 

A feltételek az if kulcsszó vagy a vele ekvivalens :- jel után állnak, és több szabály esetén logikai kapcsolatukat is meg kell jelölnünk. Ilyen logikai alapszavak és a vele ekvivalens írásjelek a következõk: 

and illetve , (vesszõ) 

or illetve ; (pontosvesszõ) 

not 

Néhány Prolog implementáció csak az egyik formát fogadja el, mások egy szabályon belül csak egyfajta jelölést engednek meg, a Turbo Prolog viszont teljes szabadságot ad a program szerkesztõinek e kérdésben. Stilisztikai szempontból azonban jó, ha egységes jelölésrendszernél maradunk, én például a kulcsszavas írásmódhoz igyekszem ragaszkodni, mert a szavak kifejezõbbek és fõleg kevésbé téveszthetõek, mint az írásjelek. 

Természetesen a feltételek végén minden esetben ott kell lennie a szabályt lezáró pontnak. 

A Prolog programok szerkezetének további boncolgatása helyett lássunk olyan feladatokat, amelyek az eddigiek alapján megoldhatóak, vagy a megoldás során vesszük elõ az újabb lehetõségeket és szintaktikai szabályokat. 

II.FEJEZET

UJJGYAKORLATOK PROLOGBAN 
(programozási tételek)

II.1. Eldöntés

Az eldöntés programozási tétele arra ad algoritmust, hogy egy adatstruktúrában van-e T tulajdonságú elem. 
A Prolog legtermészetesebb adatstruktúrája a lista, így a feladatot egy kicsit egyszerûsítve úgy fogalmazzuk át, hogy van-e a listában egy bizonyos elem? Lássunk egy kész megoldást: 

domains 

elem = symbol 
lista = elem*
predicates 

bennevan(elem,lista)
clauses 

bennevan(E,[E|_]). 
bennevan(E,[_|Lv]) if bennevan(E,Lv).
A magyarázat most még sokkal részletesebb lesz, mint késõbb, mert sok ismétlésbe nem szándékozom majd bocsátkozni. 

A domains szekció deklarál egy elem nevû, tetszõleges szimbólumból álló elemi objektumot, valamint egy lista nevû, elem típusú tagokból álló listát. 

A predicates egyetlen szabályt deklarál, amelynek a neve bennevan, két argumentummal, amelyek közül az elsõ elem típusú, a második pedig lista típusú. E deklaráció után ragaszkodnunk kell a paraméterek ilyen sorrendjéhez! 

Az a tény, hogy nincs goal, vagyis a programban meghatározott célkifejezés, azt eredményezi, hogy egy RUN parancsra megkérdezi a célt. Ezt beadva fog csak mûködni a program. 

A clauses szekció tartalmazza a bennevan szabályait. Az elsõ szabállyal nincs is semmi probléma: az elem típusú, tehát szimbólumot tartalmazó E változó benne van a listában, ha az E a lista feje, akármi is áll a lista további részében. Ha ez teljesül, akkor már nem kell semmi mást csinálni, ezt jelzi az, hogy elértük a szabály végét jelzõ pontot. Ez egy eklatáns példa az argumentumaikban kielégített szabályokra. 
Aki írt már rekurzív programot, annak a második szabály nem különösebben meglepõ: az E elem benne van a listában, ha a lista elsõn kívüli részében benne van. Ezzel a lehetséges eseteket szétválasztottuk, hiszen egy lista a lista fejébõl és a lista elsõnkívüli elemeibõl álló listából áll. A szabály nemrekurzív elsõ sora elintézi az elsõ esetet, a második, rekurzív sor pedig a másodikat. 
Procedurálisan úgy képzelhetjük el a program mûködését, hogy megnézi a lista elsõ elemét, és ha az egyenlõ a keresett szimbólummal, akkor a bennevan-ság teljesül. Ha nem, akkor keres egy újabb szabályt, ami most azt mondja ki, hogy akkor teljesül a bennevan-ság, ha a lista maradék részére az teljesül. Tehát újra kezdi a vizsgálatot, már egy rövidebb listával, és természetesen újból az elsõ szabályt próbálja alkalmazni. Ha valaha teljesül az egyezés, akkor "örömében.." visszalép a rekurzió egy magasabb szintjére, ahol most pontot talál. Ennek az a jelentése, hogy nincs további teendõ, tehát ismét visszalép, egészen addíg, amíg a parancs-szintig el nem ér. Itt kiírja, hogy True, angolul ez igaz-at jelent. Ha a sorozatos rekurzív hívások által annyira lerövidül a lista, hogy már nincs benne semmi, csak az üres lista,akkor "bánatosan" konstatálja, hogy nincs meg a keresett elem a listában, majd a rekurzív hívásokból visszalépve azt írja a képernyõre, hogy False. Szemléltetésül bemutatom a fenti program futását a 

Goal:bennevan("f",["a","b","f","c"])
céllal, a TRACE bekapcsolásakor adott visszajelzésen keresztül: 

CALL: bennevan("f",["a","b","f","d"]) 
REDO: bennevan("f",["a","b","f","d"]) 
CALL: bennevan("f",["b","f","d"]) 
REDO: bennevan("f",["b","f","d"]) 
CALL: bennevan("f",["f","d"]) 
RETURN:*bennevan("f",["f","d"]) 
RETURN: bennevan("f",["b","f","d"]) 
RETURN: bennevan("f",["a","b","f","d"]) True
Most egy olyan célállításról kérdezzük a gép véleményét, amely szemmel láthatóan kudarccal fog végzõdni: 

Goal:bennevan("z",["a","b","c"])
A nyomkövetés eredménye az alábbi lista: 

CALL: bennevan("z",["a","b","c"]) 
REDO: bennevan("z",["a","b","c"]) 
CALL: bennevan("z",["b","c"]) 
REDO: bennevan("z",["b","c"]) 
CALL: bennevan("z",["c"]) 
REDO: bennevan("z",["c"]) 
CALL: bennevan("z",[]) 
REDO: bennevan("z",[]) 
FAIL: bennevan("z",[]) False
Mindaz, amit elmondtam, követhetõ ebbõl a listából. Továbbhaladni viszont csak akkor érdemes az Olvasónak, ha a program és a futás minden lépését megértette. 

II.2. Szélsõérték-kiválasztás

Ez a tétel tulajdonképpen az adathalmazból egy minimum- vagy maximumérték kiszámítására ad algoritmust. Nézzük, hogyan valósul meg ez egy logikai nyelvben, ahol nincs ciklus-utasítás? 
Az elsõ és legegyszerûbb szabály az lehet, hogy az egyelemû lista egyetlen eleme maga szélsõérték is: 

maximum([E],E). 
illetve 
minimum([E],E).
Az alkalmazott E változó olyan típusú, amilyent a szabály e helyére deklaráltunk, értéket pedig akkor kap, ha a szabály ilyen formája valaha teljesül. Konkrétan tehát akkor, amikor olyan paraméterekkel hívjuk meg a szabályt, hogy a lista az E elembõl áll, és az elem (amit a második helyre tettünk), ugyanez az elem. A következõ két-két szabály gondoskodik az összes további esetrõl. Az második szerint a lista E feje nem a szélsõérték, mert a lista E nélküli részének F maximuma vagy minimuma nagyobb vagy egyenlõ, illetve kisebb vagy egyenklõ, mint az E. 

maximum([E|List],F) if maximum(List,F) and F >= E and !. 
illetve 
minimum([E|List],F) if minimum(List,F) and F <= E and !.
A sorok végén álló felkiáltójelekrõl még sok szó esik, most csak néhány szót a szerepérõl. A ! funkció neve cut, ami vágást, elvágást jelent, és értelme az, hogy a lehetséges megoldás keresését ezen az úton már nem engedjük meg. A harmadik szabály szerint a lista feje szélsõérték, mert az E fej nélküli lista F szélsõértéke kisebb illetve nagyobb, mint E. 

maximum([E|List],E) if maximum(List,F) and F < E. 
illetve 
minimum([E|List],E) if minimum(List,F) and F > E.
Nézzük ezek után a két szélsõérték kiszámítását végzõ programot! 

domains 

elem = symbol 
lista = elem*
predicates 

maximum(lista,elem) 
minimum(lista,elem)
clauses 

maximum([E],E). 
maximum([E|List],F) if maximum(List,F) and F >= E and !. 
maximum([E|List],E) if maximum(List,F) and F <= E. 
minimum([E],E). 
minimum([E|List],F) if minimum(List,F) and F <= E and !. 
minimum([E|List],E) if minimum(List,F) and F > E.
A teljes program két különbözõ funkcióra oszlik. Fontos formai szabály, hogy az összetartozó szabályok szigorúan egymást követõen helyezkedjenek el. Nagyon fontos ezen belül még az is, hogy a szabályok milyen sorrendben követik egymást, hiszen a már említett szekvenciális végrehajtási mód miatt az oly szokványos rekurzív hívások mindig elölrõl kezdik értékelni a szabályokat. 
Ilyen egyszerû relációk leírása esetén az esztétikus és elegáns axióma-felállítás tûnik a nyilvánvalóan leghelyesebbnek. Valóban, de az elegancia legtöbbször nem adódik azonnal, hanem sok gondolkodás gyümölcse. Prologban is lehet ugyanarra a feladatra többféle megoldás, ez általában a programozótól függ. A végeredmény eleganciája viszont már sokat sejtet a megoldás jóságáról és a programozó gyakorlatáról, de fõleg a képességérõl. Gyanítom, az én itt közreadott programjaim sem mindig a legtökéletesebbek, és lehet azoknál jobbakat írni. 
  

II.3. Összegzés

Egy listában elhelyezett számsorozat összegét kell most kiszámítanunk. Mivel itt az elemek nem akármilyen szimbólumok, hanem számok, az elem sem symbol, hanem integer vagy real lesz. Válasszuk az integer-t! 

elem = integer 
lista= elem*
A szabály alakja hasonló az elõzõekhez: 

osszegzes(lista,osszeg)
Maga a szabály rekurzív. Elsõ sora (a vészkijárat...) azt mondja, hogy egy egyelemû lista összege maga az elem értéke: 

osszegzes([E],E).
A második szabály pedig azt mondja ki, hogy egy lista összege úgy számítandó ki, hogy a lista fejének értékét hozzáadjuk a lista farkának összegéhez: 

osszegzes([Fej|Farok],Osszeg) if Osszegzes(Farok,Reszeredmeny)and 
                                 Osszeg=Fej+Reszeredmeny.
Deklaratív olvasata a szabálynak az, hogy az összegszámítás úgy teljesülhet, hogy az Osszeg a Fej és a Farokra vonatkozó részeredmény összege. A teljes program tehát: 

domains 

elem = integer 
lista = elem*
predicates 

osszegzes(lista,elem)
clauses 

osszegzes([E],E). 
osszegzes([Fej|Farok],Osszeg) if osszegzes(Farok,Reszeredmeny) and 
                                 Osszeg=Fej+Reszeredmeny.
A program nem tartalmaz célállítást, tehát a RUN parancsra egy kérdést vár. Legyen ez a 

Goal:osszegzes([1,3,5,7],Sum)
A cél most az, hogy meghatározzuk, ez az állítás milyen feltétel mellett teljesül. A Prolog rendszer azt válaszolja, hogy ez csak a Sum=16 mellett teljesül, egyben választ is adva számunkra. Ha nem változót, hanem konstanst, azaz egész értéket teszünk az összeg helyére, akkor a célállítást másképpen értelmezi a rendszer: 

Goal:osszegzes([1,2,3],6)
hatására azt vizsgálja, hogy igaz-e az állítás, azaz tényleg 6-e az összeg. Mivel ezt igaznak találja, azt írja ki, hogy True, és befejezi a program futását. Hasonlóképpen értelmezi a feladatát akkor, ha a 

Goal:osszegzes([1,2,3],5)
célállítással kerül szembe, de ekkor az állítást nem tudván soha kielégíteni, a válasza False. Nem szabad viszont az elsõ pozícióra változót helyezni, mert a szabályokban szabad, tehát nem levezethetõ változók szerepelnének. 
  

II.4. Elem törlése listából

Nem tartozik a klasszikus programozási tételek közé, de egyike a legszükségesebb és legtanulságosabb programoknak, így itt kell róla beszélnünk. 
Az elsõ és legfontosabb lépés meghatározni a reláció kívánt értelmezését. Világos, hogy a lista egy elemét törlõ programnak három argumentuma lesz: 

· a törlendõ E elem, 

· az E elemet tartalmazó L1 lista és 

· az E elemet sehol sem tartalmazó L2 lista. 

Az alkalmas reláció alakja tehát: 

torol(L1,X,L2).
Szöveges értelmezése pedig az, hogy L2 minden esetben L1, elhagyva E minden elõfordulását. A program átgondolásánál a legegyszerûbb egy speciális esetre gondolni. Gondoljuk végig a 

torol([a,b,c,b],b,L)
kérdés megoldását, amely tipikus esete az elem törlése után maradó lista megkeresésének. A válasz itt 

L=[a,c].
A program rekurzív lesz az elsõ argumentumra nézve. Vegyük elõ a procedurális gondolkodásmódunkat! Kezdjük a rekurzív résszel. 
A listákra általánosan használt rekurzív forma az [E|Lista]. Két lehetséges eset van: 

· az egyik, mikor az E fej-elem a törölni kívánt elem, 

· a másik, mikor nem az. 

Az elsõ esetben E-nek a Lista listából való rekurzív törlésének eredménye adja a kielégítõ eredményt a kérdésre. Az alkalmas szabály tehát: 

torol([E|Lista],E,Maradek]) if torol(Lista,E,Maradek).
Gondolkodásmódot váltva, a szabály deklaratív olvasata ez: E-nek [E|Lista]-ból való törlése Maradek, ha E-nek Lista-ból való törlésének eredménye Maradek. 
A feltétel, hogy a lista feje és a törlendõ elem ugyanaz, a szabály fejében lévõ E változó által biztosított, hiszen ha a szabály igaznak bizonyul, akkor azonos változónév helyén csak azonos értékek állhatnak. 
A második esetben, ahol a törlendõ elem különbözik F-tõl, azaz a lista fejétõl, egyszerû. A kívánt eredmény egy lista, amelynek a feje F és farka az elem rekurzív törlésének eredménye. A szabály: 

torol([F|Lista],E,[F|Maradek]) if E<>F and torol(Lista,E,Maradek).
A szabály deklaratív olvasata: E törlése [F|Lista]-ból [F|Maradek], ha E különbözik F-tõl és E Lista-ból való törlésének eredménye Maradek. Az elõzõ szabállyal ellentétben a feltétel, hogy a lista feje és a törölni kívánt elem különbözik, explicit módon ki van fejezve a szabály törzsében, különben az E és az F változók felvehetnék ugyanazt az értéket is. 
Az alapeset (vészkijárat...) egyértelmû. Üres listából nem lehet elemet törölni, és az elvárt eredmény szintén az üres lista. Ez adja az utolsó szabályt: 

torol([],_,[]).
Az aláhúzásjel a középsõ paraméter helyén azt jelenti, hogy itt akármi állhat, a szabály szempontjából közömbös. Így lehet kivédeni a rendszer egy hibaüzenetét, mely szerint a torol([],E,[]). szabályban az E csak egyszer szerepel, így definiálatlan. A teljes program tehát: 

domains 

elem = symbol 
lista = elem*
predicates 

torol(lista,elem,lista)
clauses 

torol([E|Lista],E,Maradek]) if torol(Lista,E,Maradek). 
torol([F|Lista],E,[F|Maradek]) if E<>F and torol(Lista,E,Maradek). 
torol([],_,[]).
Nézzük át újra a megírt programot, és próbálkozzunk alternatív megoldásokkal. 
Elhagyva a második szabályból az E<>F feltételt, egy másik variánst kapunk a törlõ programunkra. Ez a variáns kisebb megszorítást tartalmaz, értelmezése tágabb lesz, mint a minden elõfordulást törlõ teljes programnak. Például a 

torol([a,b,c,b],b,[a,c]), 
torol([a,b,c,b],b,[a,c,b]), 
torol([a,b,c,b],b,[a,b,c])

mindegyike benne van a variáns értelmezésében, tehát célként megadva a ezeket, a válasz mindig True. Mind a fenti, mind a teljes program értelmezési tartományába beletartozik az az eset, amikor a törölni kívánt elem nincs benne egyik listában sem, például a 

torol([a],b,[a])

igaz állítás. Vannak esetek, amikor ez nem kívánatos. 
  

II.5. Összefuttatás

Ez a tétel már egy kissé bonyolultabb. Adott két rendezett lista, ezekbõl kell egy olyan listát készíteni, amely továbbra is rendezett és tartalmazza a két kiinduló lista elemeit, az azonosakat persze csak egyszer. 
A program a következõ: 

domains 

elem = symbol  
lista = elem*
predicates 

osszefut(lista,lista,lista)
clauses 

osszefut([X|Xs],[Y|Ys],[X|Zs]) if X<Y and ! and osszefut([X|Xs],Ys,Zs). 
osszefut([],[],[]). 
osszefut(Xs,[],Xs). 
osszefut([],Ys,Ys).
Vegyük sorra a szabályokat! 

osszefut([X|Xs],[Y|Ys],[X|Zs]) if X<Y and ! and osszefut([X|Xs],Ys,Zs).
Vázlatosan: két nem üres lista összefuttatását úgy kapjuk meg, hogy ha az elsõnek a fej-eleme kisebb, ez lesz a fej-eleme annak az eredmény-listának, amit a 
Harmadik szabályunkban pontosan az történik, ami az elsõben, csak most a két lista szerepe felcserélõdött. 

osszefut([],[],[]). 
osszefut(Xs,[],Xs). 
osszefut([],Ys,Ys).
Az utolsó három szabály "csupán" triviális dolgokat mond. De pontosan ezek a trivialitások vetnek véget a rekurzív hívásoknak, és kezdik visszafejteni a megoldást. A szabályok szerint az üres listák összefuttatásának eredménye az üres lista, valamint egy üres lista és egy tetszõleges lista összefuttatásának eredménye maga az említett lista, akár az elsõ, akár a második helyen áll is a nemüres lista. 

II.6. Únióképzés, metszetképzés

Az únióképzés tétele szerint az adott két -azonos elemeket egyenként nem tartalmazó- halmaz elemeit úgy kell összegezni, hogy az eredményhalmazban mindkét halmaz elemei szerepeljenek, de ne legyen benne két azonos elem. A metszetképzés eredménye a közös elemeket egyetlen egyszer tartalmazó halmaz. A halmazokra nem teszünk más (pl. rendezettségi) kikötést. 
A két tétel megoldása Prologban teljesen hasonló elveken nyugszik, ezért is vonható össze, mint az alábbi programlista is mutatja: 

domains 

elem = symbol 
lista = elem*
predicates 

bennevan(elem,lista) 
nincsbenne(elem,lista) 
unio(lista,lista,lista) 
metszet(lista,lista,lista)
clauses 

bennevan(E,[E|_]). 
bennevan(E,[_|Lv]) if bennevan(E,Lv). 
nincsbenne(F,L) if not(bennevan(F,L)). 
unio([E|L1],L2,[E|L3]) if unio(L1,L2,L3) and nincsbenne(E,L3) and !. 
unio([_|L1],L2,L3) if unio(L1,L2,L3) and !. unio([],_,_). 
metszet([E|L1],L2,[E|L3]) if metszet(L1,L2,L3) and bennevan(E,L2) and !. 
metszet([_|L1],L2,L3) if metszet(Lista1,Lista2,Lista3) and !. metszet([],_,[]).
Kezdjük az elemzést a bennevan szabály ellentétével, a nincsbenne szabállyal! Ennek képzése egyszerûen a not beépített függvénnyel történhet. Ha teljesül a belsõ kifejezés, akkor hamissá válik az érték, míg a belsõ kifejezés kudarca esetén lesz kielégítve az egész szabály. Az únióképzés elsõ és legfontosabb szabálya szerint: 

unio([E|L1],L2,[E|L3]) if unio(L1,L2,L3) and nincsbenne(E,L3) and !.
azaz az elsõ lista E feje akkor lesz eleme az eredménylistának, ha van olyan L3 lista, ami az elsõ lista maradéka és az L2 lista úniója révén keletkezik, és nincs benne E az L3 listában. 
Az azonnali rekurzív hívás azt eredményezi, hogy az elsõ lista rögtön "elfogy", így végülis hátulról kezdi majd a következõ szabályok alapján felépíteni az eredménylistát. 

unio([_|L1],L2,L3) if unio(L1,L2,L3) and !.
Ez a szabály az elõzõ kudarca esetén lép életbe, és azt fejezi ki, hogy az elsõ lista feje nem lehet eleme az eredménylistának, és az únióképzést a maradékkal kell folytatni. 

unio([],_,_).
Itt van az a szabály, ami visszafordítja az elsõ két szabály rekurzív hívásait, mert azt mondja, hogy egy üres lista és egy tetszõleges lista úniója egyenlõ ugyanazzal a listával. Az metszetképzés az elõzõtõl csak nagyon kicsiben különbözik. Az elsõ szabály: 

metszet([E|L1],L2,[E|L3]) if metszet(L1,L2,L3) and bennevan(E,L2) and !.
mely szerint az E elem benne van az eredménylistában, ha benne van a másik listában is, és az eredménylista további része metszete az elsõ lista maradékának és a második "érintetlen" listának. 

metszet([_|L1],L2,L3) if metszet(Lista1,Lista2,Lista3) and !.
A metszetképzés szabálycsoportjának második sorára ugyanaz vonatkozik, mint az únióképzés megfelelõ sorára. 

metszet([],_,[]).
E harmadik sornak a magyarázata is a metszetképzés definíciójára utal, hiszen egy üres lista és bármi másnak a metszete csakis az üres lista lehet. Éppen úgy, ahogyan az únióképzésnél, ennél is itt fordul vissza a rekurzív hívások sorozata. E két tétel egyszerûsége és hasonlatossága talán az Olvasót is meggyõzte a Prolog fölényérõl más nyelvekkel szemben a tiszta és az emberi gondolkodásmódot követõ feladatmegfogalmazása által. 
  

II.7. Rendezés permutálással

A listák rendezésének logikai specifikációja az, hogy keressük a lista olyan permutációját, amely rendezett. Ez azonnal leírható, mint egy logikai program. Az alapreláció: 

rendperm(Xs|Ys)
ahol Ys az Xs elemeibõl álló növekvõen rendezett lista: 

rendperm(Xs,Ys) if permutacio(Xs,Ys) and rendezett(Ys) and !.
A felkiáltójel (cut) letiltja a további megoldások keresését, hiszen annak nincs értelme. A rendezés legfelsõ szintjét lebontottuk, a felülrõl lefelé haladó programfejlesztés szellemében. Most már csak a permutacio és a rendezett szabályokat kell definiálnunk. 
Annak eldöntése, hogy egy lista növekvõen rendezett-e, kifejezhetõ az alábbi két szabállyal. Az egyik kijelentés szerint az egyelemû lista szüségszerûen rendezett. A másik szabály szerint egy lista rendezett, ha az elsõ eleme kisebb vagy egyenlõ mint a második, és a maradék lista a második elemtõl kezdve rendezett. 

rendezett([_]). 
rendezett([X|[Y|Ys]]) if X<=Y and rendezett([Y|Ys]).
A permutálást végzõ program még érdekesebb. A permutálási eljárás egy megoldása az lehet, hogy véletlenszerûen választunk egy elemet a listából, azt a lista elejére helyezzük, majd rekurzívan permutáljuk a maradék listát. Az alapkijelentés szerint az üres lista egyetlen permutációja önmaga. 

permutacio(Xs,[Z|Zs]) if kiejt(Z,Xs,Ys) and permutacio(Ys,Zs). 
permutacio([],[]).
Szólnom kell a kiejt szabályról. Ennek szerepe az, hogy a lista elejétõl kezdve a vizsgálatot, az elsõ kívánt elemet kiemelve adja vissza a bemenõ listát. Tehát ez nem azonos a II.4. fejezet programjával, amely az összes elõfordulást kitörli, csak hasonlít rá. 
Deklaratív olvasásban így hangzik: E-t kiemelhetjük az [E|Lista] listából a Lista visszaadásával, vagy E-t kiemelhetjúk az [F|Lista1]-ból, ha megadjuk az [F|Lista2] listát, ahol a Lista2 egyenlõ az E elemnek Lista1-bõl való kiemelése eredményével. 
Érdemes most áttekinteni a teljes programot: 

domains 

elem = symbol 
lista= elem*
predicates 

permutacio(lista,lista) 
kiejt(lista,elem,lista) 
rendezett(lista) 
rendperm(lista,lista)
clauses 

rendperm(Xs,Ys) if permutacio(Xs,Ys) and rendezett(Ys) and !. 
permutacio(Xs,[Z|Zs]) if kiejt(Xs,Z,Ys) and permutacio(Ys,Zs). 
permutacio([],[]). 
kiejt([E|Lista],E,Lista). 
kiejt([F|Lista1],E,[F|Lista2]) if kiejt(Lista1,E,Lista2). 
rendezett([_]). 
rendezett([X|[Y|Ys]]) if X<=Y and rendezett([Y|Ys]).
A permutációs rendezés sajnos nem túl szerencsés megoldás, mert a permutációk elõállítása csak 1/n!-al növeli a rendezettség elérésének valószínûségét lépésrõl lépésre. A permutációk sorában "egyszer csak" felbukkan az "igazi", és ennek ugyanakkora a valószínûsége, mint bármely másik permutációnak. 

II.8. Rendezés minimumkiválasztással

A listák rendezésének egy másik módja, hogy az adott listából kiemeljük a legkisebbet és az eredménylista elejére tesszük, amelynek a "vége" már rendezett, ugyanilyen módon. 
Deklaratívan megfogalmazva ez úgy hangzik, hogy egy lista akkor rendezett, ha a lista feje kisebb a rendezett részlistában szereplõ összes elemnél. 
Ez utóbbi megközelítés ihlette a minimumkiválasztásos rendezés megvalósítását az alábbi módon: 

domains 

elem = symbol 
lista = elem*
predicates 

minimum(lista,elem) 
kiejt(lista,elem,lista) 
rendmin(lista,lista)
clauses 

minimum([E],E). 
minimum([E|Lista],F) if minimum(Lista,F), F<=E and !. 
minimum([E|Lista],E) if minimum(Lista,F) and F>E. 
kiejt([E|Lista],E,Lista) if !. 
kiejt([F|Lista1],E,[F|Lista2]) if kiejt(Lista1,E,Lista2) and !. 
kiejt([],_,[]). 
rendmin([],[]). 
rendmin(Lista1,[E|Lista2]) if minimum(Lista1,E) and kiejt(Lista1,E,Lista3) and rendmin(Lista3,Lista2) and !.
A program trükkje az, hogy kiválasztjuk az adott Lista1 E minimumát a minimum szabályokkal, majd ezt az elemet kiejtjük a Lista1 eredeti listából, hogy egy olyan Lista2 végére fûzhessük, amit már lerendeztünk. A Lista3 olyan átmeneti lista, amely már nem tartalmazza az E elemet. 
Az elsõ szabály szerint az üres lista eleve rendezett. Az említett kiejtés az elõzõ fejezetben említett kiejt eljárással történik. 

II.9. Rendezés beszúrással

A rendezési feladatnak az eddiektõl eltérõ megoldása az lehet, hogy ha adott egy rendezett lista (és az üres lista mindig adott!), akkor a kívánt E elemet úgy illesztjük bele a rendezett listába, hogy azt szétválogatjuk E-nél kisebb-egyenlõ és nagyobb részekre, majd összerakjuk a három részlistát. 
Ennek megvalósítása a következõ program: 

domains 

elem = symbol 
lista = elem*
predicates 

rendbesz(lista,lista) 
szetval(lista,elem,lista,lista) 
osszerak(lista,lista,lista)
clauses 

szetval([F|L1],E,[F|L2],L3) if F <= E and szetval(L1,E,L2,L3). 
szetval([F|L1],E,L2,[F|L3]) if F > E and szetval(L1,E,L2,L3). 
szetval([],_,[],[]). 

osszerak([E|L1],L2,[E|L3]) if osszerak(L1,L2,L3). 
osszerak([],Lista,Lista). 
rendbesz([E|L1],L2) if rendbesz(L1,L3) and szetval(L3,E,Lk,Ln)and 
                       osszerak(Lk,[E|Ln],L2) and !. 
rendbesz([],[]).
Kezdjük az ismerkedést a szetval részprogrammal, amelynek az a feladata, hogy az elsõ paraméterként megadott [F|L1] lista F elemét elhelyezze vagy a kisebb-egyenlõk közé (l.szabály), vagy az F-nél nagyobbak közé (2.szabály), attól függõen, hogy melyik reláció teljesül. Mielõtt azonban elhelyezné, egy rekurzív hívásra kényszeríti önmagát, hogy az L2 és az L3 listák már rendelkezésre álljanak. Ha az elsõ paraméter az üres lista, akkor persze nincs mit tenni, két üres listát ad vissza eredményként. 
Másik segédprogramocskánk az osszerak nevû, amely szintén ravasz módon dolgozik. Feladata az, hogy egyesítsen két listát. Az únió-képzéssel ellentétben most nem kell azt vizsgálnunk, hogy létezik-e egy elem már a listában, mert ez megengedett. Ellenben ügyelni kell a sorrendre, amit úgy old meg, hogy sorozatos rekurzív hívásokkal "kiüríti" az elsõ argumentumban adott listát, majd érvénybe lép a második 

II.10. Gyorsrendezés (Quicksort)

Az egyik leghatékonyabb rendezési algoritmus a Quicksort. Elve az, hogy egy elemet kiválasztva, kettéosztjuk a listát nála kisebb és nála nagyobb elemekbõl álló részlistákra, majd ezeket rendezzük, végül újra egyesítjük a kisebbek listáját, a kiemelt elemet és a nagyobbak listáját. 
Már az elve is rekurziót sejtet, így igazán hatékonyan csak a rekurziót megengedõ programozási nyelvekben valósulhat meg. Tulajdonképpen csak a fõ eljárás az új, azok a részprogramok, amiket használ, már ismerõsek a beszúrásos rendezésbõl. 
Maga a program az alábbi: 

domains 

elem = symbol 
lista = elem*
predicates 

rendquick(lista,lista) 
szetval(lista,elem,lista,lista) 
osszerak(lista,lista,lista)
clauses 

szetval([F|L1],E,[F|L2],L3) if F <= E and szetval(L1,E,L2,L3). 
szetval([F|L1],E,L2,[F|L3]) if F > E and szetval(L1,E,L2,L3). 
szetval([],_,[],[]). 
osszerak([E|L1],L2,[E|L3]) if osszerak(L1,L2,L3). 
osszerak([],Lista,Lista). 
rendquick([E|L1],L2) if szetval(L1,E,Kis,Nagy) and 
                        rendquick(Kis ,Kisrend) and 
                        rendquick(Nagy,Nagyrend) and 
                        osszerak(Kisrend,[E|Nagyrend],L2). 
rendquick([],[]).
Ez a program nem igényel bõvebb magyarázatot. Csak az összehasonlítás kedvéért bemutatom ugyanezt az rendezési algoritmust Pascal nyelven. 
Mint ismert, a Pascalban megengedett a rekurzió, tehát nem olyan sportszerûtlen ez az összehasonlítás, mintha például a BASIC-el vetném össze a Prolog programot. A programlista hossza és áttekinthetõsége magáért beszél. 
Adottnak tételezzük fel az egészekkel feltöltött a[1..vege] tömböt. 

procedure quick(also,felso:integer); 
  procedure quick1(also,felso:integer); 
    var 
      bal,jobb,valaszto,id : integer; 
    begin 
      bal:=also; jobb:=felso; 
      valaszto:=a[(also+felso) DIV 2]; 
      repeat 
        while a[bal] < valaszto do bal:=bal+1; 
        while a[jobb] > valaszto do jobb:=jobb-1; 
        if bal<=jobb 
          then 
            begin 
              id:=a[bal]; a[bal]:=a[jobb]; a[jobb]:=id; bal:=bal+1; jobb:=jobb-1 
            end; 
      until bal>jobb; 
      if also < jobb then quick1(also,jobb); 
      if bal < felso then quick1(bal,felso); 
    end; 
  begin 
    quick1(1,vege) 
  end;
III. fejezet

Logika logikai nyelven

III.1. Logikai áramkörök

Minden logikai áramkör leírható Prolog függvénnyel, ahol a függvény az áramkör bemenõ (INPUT) pontjai és kimenõ (OUTPUT) pontjai közötti kapcsolatot írja le. 
Az alapáramköröket az igazságtáblázatukkal adjuk meg. Alkalmas lehet például egy adott kapcsolás ellenõrzésére, ha megadjuk a kívánt igazságtáblázatot és összevetjük egy alapáramkörökbõl felépített kapcsolás mûködésével, vagy fordítva, egy adott kapcsolás igazságtáblázatának elkészítésére is alkalmazhatjuk az ilyen programokat. 
A közölt kis program csak bemutatja, hogyan épülhet fel alapáramkörökbõl (elemi logikai függvényekbõl) egy másik logikai áramkör (függvény). 
Alapkapcsolásnak tekintjük a nem , az es és a vagy függvényeket, amelyek igazságtáblázatát tényekként fel kell sorolnunk a clauses szekcióban. 
A következõ program a KIZÁRÓ-VAGY, a NEM-ÉS és a NEM-VAGY logikai áramköröket állítja elõ: 

domains 

d = i ; h
predicates 

nem(d,d) 
es(d,d,d) 
vagy (d,d,d) 
xor (d,d,d) 
nand(d,d,d) 
nor (d,d,d)
clauses 

nem(i,h). 
nem(h,i). 
es(i,i,i). 
es(i,h,h). 
es(h,i,h). 
es(h,h,h). 
vagy(i,i,i). 
vagy(i,h,i). 
vagy(h,i,i). 
vagy(h,h,h). 
xor(Input1,Input2,Output) if nem(Input1,N1) and 
                             nem(Input2,N2) and 
                             es(Input1,N2,N3) and 
                             es(Input2,N1,N4) and 
                             vagy(N3,N4,Output). 
nand(Input1,Input2,Output) if es(Input1,Input2,N1) and nem(N1,Output). 
nor (Input1,Input2,Output) if vagy(Input1,Input2,N1) and nem(N1,Output).
A d adattípus csak az i és a h értékeket veheti fel. A not például akkor igaz, ha az elsõ és a második paraméter egymástól különbözik. Tehát ha az egyiket megadjuk, akkor a másik helyén lévõ érték annak ellenkezõje lesz. 
Ugyanígy az es függvény csak akkor ad i igaz értéket, ha mind az elsõ, mind a második paraméter i . Akármelyik paraméter helyébe változót téve, megkeresi azt a mintát, ami megegyezik a megadottal, egyúttal a változó felveszi a mintából vett értéket. 
Itt tehát azonnal és nagyon szemléletes módon jelentkezik a Prolog rendszer alapvetõ feladatmegoldási stratégiája, a mintaillesztés. 
RUN parancsra a programnak célként meg lehet adni bármelyik logikai áramkör bemenõ állapotát, az eredmény az OUTPUT ponton jelentkezõ érték lesz: 
  

	Goal:xor(i,h,Z)   
Z=i   
1 Solution   

Goal:nand(i,i,Z)   
Z=h   
1 Solution   

Goal:nor(h,i,Z)   
Z=h   
1 Solution


Tehát a Z helyére i-t helyettesítve teljesülhet az xor szabálya, és összesen 1 megoldás kapható. 
Hasonlóan értelmezendõ a többi kérdés is. A KIZÁRÓ-VAGY blokkrajza az alábbi, bejelölve az ideiglenesen bevezetett változók érvényességi szakaszait: 

[image: image1]
A másik két áramkört az egyszerûsége miatt nem rajzolom meg. Nagyobb programok sokkal bonyolultabb áramköröket szimulálhatnak, sõt akár ki is fejleszthetnek kívánt kapcsolásokat. 
Ez a feladat a mesterséges intelligencia területére vinne már, most viszont más irányokat kívánunk felderíteni, így itt az ideje, hogy új feladat után nézzünk. 
  

III.2. Ki a gyilkos ?

Maigret felügyelõt egy éjszakai gyilkosság helyszínére hívták, ahol már az összes szóba jöhetõ személyt kihallgattak, a nyomokat rögzítették, a tényeket és a vallomásokat összegezték. Éppen csak az maradt hátra, hogy a gyilkos személyét kiderítsék. A felügyelõ ránézett a vizsgálat eredményeire, és rámutatott az egyik szereplõre. Az nagyot nézett, majd vonakodva bár, de beismerte bûnösségét. 
Maigret beosztottjait furdalta a kíváncsiság, honnan jött rá a fõnökük a megoldásra. Végül abban állapodtak meg, hogy Prolog programot írnak az esetre, és így ellenõrzik a híres nyomozót. Sorra vették a tényeket és a háttérinformációkat, majd következtetéseket írtak le a gyanúsítottak személyére, végül a gyilkosra. A Prolog rendszerük persze teljes mértékben igazolta Maigret megállapítását... 
Nézzük meg mi is, hogyan nézett ki a beosztottak programja? 

domains 

neve, neme, foglalkozasa, targy, bun, anyag = symbol 
kora=integer
predicates 

szemely(neve, kora, neme, foglalkozasa) 
viszonya_volt(neve, neve) 
meggyilkoltak(neve, targy) 
megolve(neve) 
gyilkos(neve) 
inditek(bun) 
bepiszkolva(neve, anyag) 
van_neki(neve, targy) 
ugyanazt_okozza(targy, targy) 
tulajdona_lehet(neve, targy) 
gyanus(neve)
clauses 

/*** A gyilkosság körülményei ***/ 
szemely(bert , 55, ferfi, acs). 
szemely(allan, 25, ferfi, focista). 
szemely(allan, 25, ferfi, hentes). 
szemely(john , 25, ferfi, zsebtolvaj). 
viszonya_volt(barbara, john). 
viszonya_volt(barbara, bert). 
viszonya_volt(susan , john). 
meggyilkoltak(susan, golfuto). 
megolve(susan). 
inditek(penz). 
inditek(feltekenyseg). 
inditek(igazsagtalansag). 
bepiszkolva(bert , ver). 
bepiszkolva(susan , ver). 
bepiszkolva(allan , sar). 
bepiszkolva(john , csoki). 
bepiszkolva(barbara, csoki). 
van_neki(bert, setabot). 
van_neki(john, pisztoly). 

/*** Háttérinformációk ***/ 
ugyanazt_okozza(setabot , golfuto). 
ugyanazt_okozza(karo , golfuto). 
ugyanazt_okozza(ollo , kes). 
ugyanazt_okozza(futballcipo, golfuto). 
tulajdona_lehet(X, futballcipo) if szemely(X, _, _, focista). 
tulajdona_lehet(X, ollo) if szemely(X, _, _, fodrasz). 
tulajdona_lehet(X, Targy) if van_neki(X, Targy). 

/* Gyanusak azok, akiknek van olyan fegyverük, amivel megölték Susan-t*/ 
gyanus(X) if meggyilkoltak(susan, Fegyver) and 
             ugyanazt_okozza(Targy, Fegyver) and 
             tulajdona_lehet(X, Targy). 

/* Gyanusak azok, akiknek viszonya volt Susan-al */ 
gyanus(X) if inditek(feltekenyseg) and 
             szemely(X, _, ferfi, _) and 
             viszonya_volt(susan, X). 

/*Gyanusak azok a nõk, akiknek viszonya volt Susan ismerõseivel*/ 
gyanus(X) if inditek(feltekenyseg) and 
             szemely(X, _, no, _) and 
             viszonya_volt(X, Ferfi) and 
             viszonya_volt(susan, Ferfi). 

/* Gyanusak a zsebtolvajok, akiket a pénz motívál. */ 
gyanus(X) if inditek(penz) and 
             szemely(X, _, _, zsebtolvaj). 

/*Gyilkos az a személy, aki gyanús és olyan nyomok vannak rajta, amilyenek az áldozaton voltak.*/ 
gyilkos(Gyilkos) if szemely(Gyilkos, _, _, _) and 
                    megolve(Megolve) and 
                    Megolve <> Gyilkos /* Mert nem öngyilkosság */ and 
                    gyanus(Gyilkos) and 
                    bepiszkolva(Gyilkos, Nyom) and 
                    bepiszkolva(Megolve, Nyom).
 

Megfigyelhetõ, hogy a nyomozás eredményei közül nem mindegyik játszik szerepet a megoldás kialakításában, tehát van "fölösleges" információ. Ez a redundancia azonban az ilyen feladatokban természetes. Egy Prolog program írásánál azonban igyekezzünk olyan információkat felhasználni, amik szükségesek és elegendõek. 
Az olyan programok, amik pontosan a kellõ számú és tartalmú szabályokat és tényeket tartalmazzák, nevezzük teljes programoknak. Ezzel szemben vannak "bõ" és "szûk" programok, mint például ez a feladat is. 
  

III.3. A tolvaj szarka

A feladat Bizám György-Herczeg János "Sokszínû logika" címû könyvének 39. és 40. feladata, és tulajdonképpen diofantoszi egyenlet megoldását jelenti. Hogy a könyvben leírt megoldás ennél ravaszabb, sokkal inkább a logikai megfontolásokra épül, az bizonyára azért van, mert így egyszerûbb mint a mechanikus, de fáradságos egyenletmegoldás. 
A számítógép nagy sebessége azonban más megoldásra sarkallja a felhasználót. Dolgozhat a gép sokat -az ésszerûség határain belül-, csak az emberi munka csökkenjen lényegesen! 
Erre mutatok egy megoldást, ami programozás szempontjából roppant egyszerû, ugyanakkor a géptõl sok próbálgatást jelent, míg végül a jó megoldást megkapjuk. 
A feladat szövege szerint egy szarka aprópénzt lopkodott össze, összesen 10 darab érmét (10,20,50 filléreseket és 1,2,5,10 forintosakat értve ezek alatt) 13.40 Ft értékben. A kérdés, hogy milyen érmékbõl hányat gyûjtött? 
A módszer az, hogy módszeres próbálgatással kiválasztunk 10 érmét, és ha összértékük 13.40 Ft, akkor az egy jó megoldás. Az ezt megvalósító szabály tehát: 

penz(P,[F10,F20,F50,E1,E2,E5,E10]) if 
     darab([F10,F20,F50,E1,E2,E5,E10]) and 
     X=0.1*F10+0.2*F20+0.5*F50+E1+2*E2+5*E5+10*E10 and 
     abs(P-X)<0.05.
P az összeg értéke, Fxx és Exx a fillérek illetve a forintok darabszáma. Az egyenlõség szigorú elvárása helyett a gépi ábrázolás pontatlansága miatt a számított X érték és a kívánt P különbségének megfelelõ kis érték alá esését követeljük meg. A darabszám szabályai a feladat szövegébõl adódnak. 
A filléres és 1, 2 forintos érmék száma a 0,1,2,3,4,5 számok valamelyike lehet, míg az 5 és 10 forintosok kevesebb számban fordulhatnak elõ: 

darab([F10,F20,F50,E1,E2,E5,E10]) if 
    kiemel([0,1,2,3,4,5],F10) and 
    kiemel([0,1,2,3,4,5],F20) and 
    kiemel([0,1,2,3,4,5],F50) and 
    kiemel([0,1,2,3,4,5],E1) and 
    kiemel([0,1,2,3,4,5],E2) and 
    kiemel([0,1,2],E5) and 
    kiemel([0,1],E10) and 
    F10 + F20 + F50 = 5 and 
    E1 + E2 + E5 + E10 = 5.
Itt szerepel az a szabály is, hogy a filléresek száma és a forintosok száma egyaránt 5. Csak ha ezek mindegyike teljesül, akkor teljesül a darab eljárás. A kiemel szabálya az adott listából kiemel egy értéket. Rekurzív volta és a rendszerszintû backtrack keresõ algoritmus miatt lesz mindíg más a darabszámok értéke, így végül az összes lehetséges esetet meg fogja vizsgálni. 
A teljes lista pedig így alakul: 

domains 

lista= integer
predicates 

penz(real,lista) 
darab(lista) 
kiemel(lista,integer)
clauses 

penz(P,[F10,F20,F50,E1,E2,E5,E10]) if 
    darab([F10,F20,F50,E1,E2,E5,E10]) and 
    X=0.1*F10+0.2*F20+0.5*F50+E1+2*E2+5*E5+10*E10 and 
    abs(P-X)<0.05. 
darab([F10,F20,F50,E1,E2,E5,E10]) if 
    kiemel([0,1,2,3,4,5],F10) and 
    kiemel([0,1,2,3,4,5],F20) and 
    kiemel([0,1,2,3,4,5],F50) and 
    kiemel([0,1,2,3,4,5],E1) and 
    kiemel([0,1,2,3,4,5],E2) and 
    kiemel([0,1,2],E5) and 
    kiemel([0,1],E10) and 
    F10 + F20 + F50 = 5 and 
    E1 + E2 + E5 + E10 = 5. 

kiemel([X|_],X). 
kiemel([_|Xs],X) if kiemel(Xs,X).
A törtszámok elõfordulása miatt lesz a penz elsõ paramétere valós típusú, míg az érmék számai egész értékek. A program futtatása és eredménye: 
  

	Goal:penz(13.4,[F10,F20,F50,E1,E2,E5,E10])   
F10=2, F20=1, F50=2, E1=1, E2=3, E5=1, E10=0   
1 Solution


A következõ feladat ennek apró módosítása, ami a könyv módszerének teljes újragondolásával jár, míg a Prolog programot csak kis mértékben kell átalakítanunk. Csupán az összeg változott meg 15 forintra, az érmék száma 11-re, a fillérek száma 6-ra és két új szabály: egyforintos több volt, mint ötös, és hogy kettes nem kevesebb, mint tízes. 
Csak a penz és a darab szabályokat adom meg: 

penz(P,[F10,F20,F50,E1,E2,E5,E10]) if 
    darab([F10,F20,F50,E1,E2,E5,E10]) and 
    X=0.1*F10+0.2*F20+0.5*F50+E1+2*E2+5*E5+10*E10 and 
    abs(P-X)$<$0.05. 
darab([F10,F20,F50,E1,E2,E5,E10]) if 
    kiemel([0,1,2,3,4,5,6],F10) and 
    kiemel([0,1,2,3,4,5,6],F20) and 
    kiemel([0,1,2,3,4,5,6],F50) and 
    kiemel([0,1,2,3,4,5],E1) and 
    kiemel([0,1,2,3,4,5],E2) and 
    kiemel([0,1,2],E5) and 
    kiemel([0,1],E10) and 
    E1 > E5 and 
    E10 <= E2 and 
    F10 + F20 + F50 = 6 and 
    E1 + E2 + E5 + E10 = 5.
Ez a változat a következõ futtatási képet mutatja: 
  

	Goal:penz(15,[F10,F20,F50,E1,E2,E5,E10])  
F10=1, F20=2, F50=3, E1=3, E2=0, E5=2, E10=0  
1 Solution 


  
  
  

III.4. Hanoi tornyai

Ez egy klasszikus logikai feladat, bizonyára közismert is. Van három rúd, amelyekre korongokat helyezhetünk. A feladat az, hogy az egyik rúdról átrakjuk az összes korongot a másikra egy harmadik segítségével, az alábbi szabályok betartása mellett: 

· egyszerre csak egy korongot mozgathatunk 

· kisebb korong nem kerülhet nagyobb alá 

A megoldás elõtt gondoljuk végig a játék jellemzõit. Elõszöris ez egy olyan feladat, amit jó taktikával -azaz a helyes módszerrel- felesleges próbálgatás nélkül meg lehet oldani. Vezessük ugyanis vissza a feladatot eggyel kevesebb számú korong esetére, és ha azt megoldottuk, akkor már könnyen adódik a megoldás. Tehát ez a feladat tisztán rekurziós jellegû! 
Hogy miben áll a rekurzió, az alábbiakból kiderül. Van N korong a kiindulási rúdon, ezt kell a célrúdra átraknunk. Ha N-1 korongot átrakunk a szabad rúdra és az utolsó korongot átrakjuk a célrúdra, akkor már csak az N-1 korongot kell áthelyezni a szabad rúdról a célrúdra. A rekurziót tehát kétszer kell alkalmazni. 
Abban az esetben, ha csak egyetlen korongot kell áthelyezni, akkor nincs szükség rekurzív hívásra, egyszerûen át kell rakni azt és kész. 
Bármilyen meglepõ, de ezeknek a mondatoknak Prolog nyelvre fordítása már elegendõ a rendszernek ahhoz, hogy megoldja a problémát. Annyival azonban tanácsos kiegészíteni, hogy miközben a korongokat rakosgatja, informáljon bennünket a képernyõn. Most lássuk a Prolog program indítását és futásának eredményét, és utána elemezzük a programot részletesen. 
  

	Goal:hanoi(3)   
Tegyél egy korongot a baloldali rúdról a jobboldali rúdra.   
Tegyél egy korongot a baloldali rúdról a kozepso rúdra.   
Tegyél egy korongot a jobboldali rúdról a kozepso rúdra.   
Tegyél egy korongot a baloldali rúdról a jobboldali rúdra.   
Tegyél egy korongot a kozepso rúdról a baloldali rúdra.   
Tegyél egy korongot a kozepso rúdról a jobboldali rúdra.   
Tegyél egy korongot a baloldali rúdról a jobboldali rúdra.  
True


A program listája az alábbi: 

domains 

rud = jobboldali ; kozepso ; baloldali
predicates 

hanoi(integer) 
tedd(integer,rud,rud,rud) 
inform(rud,rud)
clauses 

hanoi(N) if tedd(N,baloldali,kozepso,jobboldali). 

tedd(1,A,_,C) if inform(A,C) and !. 
tedd(N,A,B,C) if N1=N-1 and 
                 tedd(N1,A,C,B) and 
                 inform(A,C) and 
                 tedd(N1,B,A,C). 

inform(Rud1,Rud2) if 
    write("\nTegyél egy korongot a ",Rud1, " rúdról a ",Rud2, " rúdra").
Az elsõ szembetûnõ dolog a domains szekcióban van, aholis a rud típus három alternatív konstans értékeket vehet fel, úgymint baloldali, kozepso, jobboldali. Ahol a predicates részben a rud típus szerepel, ott csak ez a három érték fordulhat elõ. 
A használt szabályok közül az elsõ a hanoi(N). Ezt kell célként megadni a RUN parancs után, ahol N a korongok száma. Funkciója csak annyi, hogy közvetítsen a további szabályok felé, így iteratív szabályként veendõ. Magában foglalja azt, hogy a baloldali rúdról a kozepso rúd felhasználásával a jobboldali rúdra kívánjuk áthelyezni az N darab korongot. 
Az inform(rud,rud) szabály egy olyan eljárás, amely kiír egy olyan kerek kis mondatot, ami információt ad a teendõrõl. Az itt használt write beépített eljárás szabályai hasonlatosak a C nyelv ilyen nevû eljárásához, így például a "\n" soremelést idéz elõ, a kiírandókat vesszõ választja el, az idézõjelbe tett string változatlanul kerül kiíratásra, hacsak nem a backslash ("\") mögött áll. 
A leglényegesebb formák a közölt programokból elsajátíthatók. A feladat érdemi részéhez értünk el. A tedd(integer,rud,rud,rud) szabályt úgy értelmezzük, hogy az elsõ paraméterként adott számú korongot kell a második paraméterként megadott rúdról a negyedik paraméterrel megadott rúdra helyezni, a harmadik paraméterben meghatározott rúd segítségével. 
Ha a korongok száma helyén 1 van, akkor a dolog egyszerû: 

tedd(1,A,_,C) if inform(A,C) and ! .
Szavakkal: tedd az A rúdról a C rúdra a korongot azzal, hogy kiírod ezt a tevékenységet. A szabály rekurzív része így néz ki: 

tedd(N,A,B,C) if N1=N-1 and 
                 tedd(N1,A,C,B) and 
                 inform(A,C) and 
                 tedd(N1,B,A,C) .
Az elsõ teendõje az, hogy egy koronggal kevesebbre állítja a paramétert, hogy elõkészítse a rekurziókat. Ha ezt nem tenné, nem vezetné vissza egyszerûbb esetekre, és soha nem érne véget a program. Utána az N1=N-1 korongot áthelyeztetjük az ideiglenes rúdra (ez tulajdonképpen ugyanennek az eljárásnak rekurzív hívása), majd ki kell írni az N-edik korong áthelyezését az A rúdról a C rúdra. Ezt közli az inform(A,C). Miután ez megtörtént, már csak az N1 korong ráhelyezése (rekurzió!) a feladat a B-vel jelzett rúdról a C rúdra, a már szabad A rúd segítségével. 
A rekurziós hívások befejeztével a kimeneten a teendõk elolvashatók, a feladatot megoldottuk. 

III/5. A becsapott edzõ

Ezt a feladatot is a "Sokszínû logika" címû könyvben találtam. Lényege, hogy öt úszó versenyének végén maguk a versenyzõk az alábbi kijelentéseket teszik: 

· András: Dezsõ végzett a második helyen, én a harmadik lettem. 

· Béla: Én értem el a legjobb eredményt. Második Csaba lett. 

· Csaba: Harmadik lettem, Béla pedig utolsó. 

· Dezsõ: Második helyen végeztem. Ernõ erejébõl a negyedik helyre futotta. 

· Ernõ: Egyetlen úszót sikerült megelõznöm. A versenyt András nyerte. 

· Edzõjük elképedt ábrázatát látva bevallották neki, hogy megtréfálták: mindegyiküknek a két információja közül csak az egyik igaz, a másik hamis. Holtverseny nem volt. 

A probléma "Prologért kiált". Addíg kell a névsort permutálni, amíg az állításoknak meg nem felelõ névsort nem kapunk. Ha nincs ellentmondás, akkor megoldottuk a feladatot. A könyvben szereplõ megoldás persze nem ilyen vaktában kezdi el a feladat elemzését, hiszen az emberi logika sokkal "lustább", mintsem hogy felesleges eseteket megvizsgáljon, másrészt egy szerencsésen megválasztott kiindulópontból nagyon gyorsan el lehet jutni a megoldáshoz. 
A Prolog nyelvû megoldás nem épít a szerencsére, de a szereplõk kis száma miatt viszonylag nem sok eset fordulhat elõ (maximum 5!=120), így a program futási ideje egyáltalán nem számottevõ. 
A problémát a megoldas szabály adja, melynek két argumentuma egy-egy lista, bennük a kiinduló és az eredményül kapott sorrenddel. Világos, hogy az utóbbi az elõzõnek egy permutációja. Ezen helyezések kialakulásához figyelembe kell venni az öt úszó öt kijelentéspárját úgy, hogy közülük csak az egyik igaz. A megoldas tehát a következõ szabály kielégítése révén adódhat: 

megoldas(K,[_1,_2,_3,_4,_5]) if 
    permutacio(K,[_1,_2,_3,_4,_5]) and 
    egyikigaz(dezso,_2,andras,_3) and 
    egyikigaz(bela, _1,csaba, _2) and 
    egyikigaz(csaba,_3,bela, _5) and 
    egyikigaz(dezso,_2,erno, _4) and 
    egyikigaz(erno, _4,andras,_1).
Hogy megoldjuk a feladatot, a szöveg tartalma szerint a kijelentéspárokon belül az egyiket igaznak, a másikat hamisnak kell feltételeznünk, ezért az egyikigaz elnevezés. Ennek négy argumentuma a kijelentéspárok jelentéstartalma (például az elsõ Dezsõ-rõl azt állítja, hogy második, illetve hogy András a harmadik). 
A feladat elemzése után azt a következtetést kell levonnunk, hogy azon túl, hogy két kijelentés egy párat alkot, mindegy, hogy azt ki mondja. Az egyikigaz szabályai a következõk 

egyikigaz(R,S,T,U) if S=R and U<>T. 
egyikigaz(R,S,T,U) if S<>R and U=T.
Az elsõ szabály szerint tehát az ismeretlen S változó megkapja az R értékét és így az elsõ kijelentés igaz, ugyanakkor T értéke nem lehet U-val egyenlõ. A második szabály szerint mindez fordítva történik, tehát a második argumentumpár lesz igaz, és az elsõ pár a hamis. 
Alkalmaznunk kell ezeken kívül a permutálást végzõ szabályokat, amelyeket az elõzõ fejezetben már ismertettem. A változatosság kedvéért a programban szerepel a célkifejezés is. 
Itt a clearwindow beépített eljárás a használt ablak törlésére szolgál. Meghívva a megoldas eljárást az ismert névsorral és az ismeretlen M listanévvel, a program igyekszik értéket találni ennek a változónak. 
A már ismertetett szabályok szerint ez a mi kívánságunkat kielégítõ megoldás lesz - ha van. Ha több megoldás lenne, a fail szó nélkül nem szereznénk róla tudomást. Mivel azonban ott van, a Prolog visszalép az utolsó elágazásra, és tovább keresgél. Ha talál, akkor azt is kiírja, ha nem, akkor befejezi a futását. 
Lássuk hát a teljes programlistát! 

domains 

uszo = symbol 
lista = uszo*
predicates 

megoldas (lista,lista) 
egyikigaz (uszo,uszo,uszo,uszo) 
permutacio (lista,lista) 
kiemel (uszo,lista,lista)
goal 

clearwindow and 
write(" A becsapott edzõ\n\n") and 
write("A lehetséges sorrend:") and 
megoldas([andras,bela,csaba,dezso,erno],M) and 
write("\n",M) and fail.
clauses 

megoldas(K,[_1,_2,_3,_4,_5]) if 
    permutacio(K,[_1,_2,_3,_4,_5]) and 
    egyikigaz(dezso,_2,andras,_3) and 
    egyikigaz(bela, _1,csaba, _2) and 
    egyikigaz(csaba,_3,bela, _5) and 
    egyikigaz(dezso,_2,erno, _4) and 
    egyikigaz(erno, _4,andras,_1). 

egyikigaz(R,S,T,U) if S=R and U<>T. 
egyikigaz(R,S,T,U) if S<>R and U=T. 

permutacio(Xs,[Z|Zs]) if kiemel(Z,Xs,Ys) and permutacio(Ys,Zs). 
permutacio([],[]). 

kiemel(X,[X|Xs],Xs). 
kiemel(X,[Y|Ys],[Y|Zs]) if kiemel(X,Ys,Zs).
Nem udvariatlanságból írtam kisbetûkkel a neveket, hanem azért, mert a Prologban nagybetûkkel csak változónév kezdõdhet, míg a konstansok a kisbetûsek. A futtatás során megjelenõ kép a következõ: 
  

	A becsapott edzõ  
A lehetséges sorrend:  
M=["dezso","csaba","andras","erno","bela"] 


A könyvben a következõ feladat ennek egy apró módosítása. Ha az utolsó kijelentés nem az, hogy "A versenyt András nyerte.", hanem az, hogy "Csaba a harmadik", akkor a programunkon csak egy kis módosítást kell végeznünk, és futtatva máris megkapjuk az eredményt: 
  

	A becsapott edzõ  
A lehetséges sorrend:  
M=["dezso","csaba","andras","erno","bela"]  
M=["bela","dezso","csaba","andras","erno"]


Láthatóan két megoldás született, és valóban, mindkettõ helyes. 

III/6. A ravasz patikus

Gyakori iskolai feladat az, hogy adott térfogatú edényekkel kell olyan térfogatot kimérni, ami nem egyezik egyik edényével sem. Képzeljük magunkat egy patikus helyébe, akinek a következõ feladatot adják: 

Van egy 8 literes és egy 5 literes edényünk, és egy nagy kád, amibõl bõven meríthetünk vizet. A feladat az, hogy az egyik edényben legyen 4 liter víz. Az edényeken nincs mérõskála. Hogyan kell öntögetni, hogy pontosan sikerüljön a dolog?
A feladat elemzése során arra a következtetésre juthatunk, hogy ezt a feladatot nem tudjuk úgy megfogalmazni, hogy a gép számára a megoldáshoz közelebbinek tûnjön az új állapot. Arra is vigyázni kell, nehogy olyan helyzet jöjjön újból elõ, amely már volt, például a kezdõállapotok egyike. Ilyen szabályok sorát kell most felállítanunk. 
Az elsõ szabály a patikus tevékenységének leírása: 

patikus( _, 4,[]). 
patikus(BR,CR,[L|LL]) if ont(BR,CR,BU,CU,L) and 
patikus(BU,CU,LL) and ! .
Tehát a patikus addíg dolgozik, amíg olyan állapot nem jön elõ, hogy a második edényben 4 liter van. Ha ez nem teljesül, akkor önt egyet, majd újból kezdi a tevékenységet (vizsgál, és ha nem a kívánt állapot, akkor önt), miközben jegyzeteli az öntögetéseit. 
A következõ szabályok (lásd a program listáját!) sorra a következõ eseteket jelentik: 

1. átöntés az 5 literesbõl a 8 literesbe, de marad még az 5-ösben 

2. átöntés az 5 literesbõl a 8 literesbe, nem marad az 5-ösben 

3. átöntés a 8 literesbõl az 5 literesbe, nem marad a 8-asban 

4. átöntés a 8 literesbõl az 5 literesbe, de marad még a 8-asban 

5. a 8 literes teletöltése, ha az 5-ös nincs tele 

6. az 5 literes teletöltése, ha a 8-as nincs tele 

7. a 8 literes ürítése, ha az 5 literes nem üres 

8. az 5 literes ürítése, ha a 8 literes nem üres 

A felsorolt esetek az összes értelmes lehetséges állapotot illetve mûveletet lefedik, így a programnak csak a megfelelõket kell kiválogatnia, hogy eljusson a kívánt célig. Az említett jegyzetelést olyan módon teszi, hogy egy listába fûzi az öntögetések utolsó paramétereit, ami végül kiírásra kerül, innen leolvasható a teendõk sora. 
A program listájában a kulcsszavak helyett a rövidebb, írásjeles írásmód is megjelenik. Remélem, így is könnyen érthetõek az alábbiak: 

domains 
terf = integer 
lista = symbol* 
predicates 
patikus(terf,terf,lista) 
ont(terf,terf,terf,terf,symbol) 
clauses 
patikus( _, 4,[]). 
patikus(BR,CR,[L|LL]) if ont(BR,CR,BU,CU,L) and 
patikus(BU,CU,LL) and ! . 

ont(BR,CR,BU, 0,_58u) if CR$>$0 , BR$<$8 , BU=BR+CR , BU$<$8. /*1*/ 
ont(BR,CR, 8,CU,_58m) if CR$>$0 , BR$<$8 , CU=BR+CR-8 , CU$>$5. /*2*/ 
ont(BR,CR,BU, 5,_85m) if BR$>$0 , CR$>$3 , BU=BR+CR-5 , BU$>$3. /*3*/ 
ont(BR,CR, 0,CU,_85u) if BR$>$3 , CR$<$5 , CU=BR+CR , CU$<$5. /*4*/ 
ont( 0, B, 8, B,tele8) if B$<$5. /*5*/ 
ont( A, 0, A, 5,tele5) if A$<$8. /*6*/ 
ont( A, B, 0, B,urit8) if A$>$0 , B$>$0 , B$<$5. /*7*/ 
ont( A, B, A, 0,urit5) if A$>$0 , B$>$0 , A$<$8. /*8*/
  

	Goal : patikus(0,0,H)  
H=["tele5","u58","tele5","m58","urit8","u58","tele5","u58","tele5","m58"]  
1 Solution


Itt a patikus-program futtatási eredménye látható. Tegyük hát sorra a teendõket: 
  

	tele5
	öntsük tele az 5 literest 

	u58
	öntsük át az 5 literesbõl az egészet a 8 literesbe

	tele5
	öntsük tele az 5 literest

	m58
	öntsünk az 5 literesbõl a 8 literesbe annyit, amennyit lehet (3 litert)

	urit8
	ürítsük ki a 8 literest

	u58
	öntsük át az 5 literesbõl az egészet a 8 literesbe

	tele5
	öntsük tele az 5 literest

	u58
	öntsük át az 5 literesbõl az egészet a 8 literesbe

	tele5
	öntsük tele az 5 literest

	m58
	öntsünk az 5 literesbõl a 8 literesbe annyit, amennyit lehet (1 litert)


  
Ezek után az 5 literes edényben 4 liter marad, a feladatot tehát megoldottuk. 

III.7. A kecske, a káposzta és a farkas

Nem ismeretlen ez a feladat sem a logikai feladatok kedvelõi elõtt. A fõszereplõ a csónakos, akinek a folyó túlsó partjára kell átvinnie egy farkast, egy kecskét és egy káposztát. Amíg a csónakos a közelben van, addíg az állatok "rendesen" viselkednek. Ha azonban nincs aki vigyázzon rájuk, akkor a farkas nyomban megeszi a kecskét, vagy a kecske a káposztát, viszont a farkas nem szereti a káposztát. A csónakban csak egyetlen utas utazhat a csónakoson kívül, de megendett a mindkét irányú szállítás, sõt az utas nélküli mozgás is. Hogyan tervezze meg az átszállítást ez a csónakos? 
Annak ellenére, hogy ez a feladat látszólag egyszerû, az alábbi program meglehetõsen hosszú. Ennek vagy az az oka, hogy nem találtam meg a legegyszerûbb módját, vagy az, hogy az emberi gondolkodásmód ilyen sokszempontú feladatok esetében még mindig sokkal hatékonyabb, mint a számítógép. 
A nehézséget az okozza, hogy le kell írnunk mindazokat a szabályokat, amiket az ember a feladat megoldása közben -tudatosan és nem tudatosan- alkalmaz. A bevezetõben a Prolog nyelv specifikációs eszközként való alkalmasságára utaltam, és itt példákat is láthatunk majd ennek bizonyítására. 
Ha a top-down módszert alkalmazzuk és procedurálisan gondolkodunk, akkor a következõ megállapításokra juthatunk: 

· A csónaknak páratlan számszor kell utaznia. Ezek a B-J irányú utazások. Minden ilyen út után meg kell vizsgálni, elértük-e a kívánt végeredményt? 

· Ha nem, akkor olyan megoldással folytatódjék a program, ami J-B irányú mozgást jelent és ezután a kívánt végcél érdekében folytatódjon egy B-J irányú úttal. 

· Csak olyan utazást lehet megengedni, ami egy még nem létezõ esetet jelent. Ha nem így lenne, az esetleg végtelen ciklust okozna. Ezért tárolni kell a kezdeti és a közbülsõ állapotokat. 

· Be kell tartanunk azt a szabályt, hogy bizonyos állapotok tiltottak (pl. farkas és kecske együtt, õrizetlenül). 

Ezek után a programról részletekben, mielõtt a teljes listát megmutatnám. A magyarázat részletes lesz, hiszen összetett programról van szó, és tanulságos, mivel megítélésem szerint több fogást érdemes lehet máshol is alkalmazni. 
Jelöljük a feladat szereplõit a programban rendre csonakos, farkas, kaposzta, kecske azonosítókkal. 
Az ures azt jelentse, hogy mindkét parton elképzelhetõ az, hogy nincs ott senki (a jobbparton ez a kiindulási állapot, a balparton pedig ez a célállapot). A csónakban is "utazhat" az "ures", azaz üres az utas helye, csak a csónakos megy át a túlpartra. 
Az ezekbõl alkotott sorozatok neve legyen lista. A domains szekció harmadik sorában szereplõ eset deklarációjára késõbb kell visszatérnünk. 

domains 

csonakos,farkas,kecske,kaposzta,ures = symbol 
lista = symbol* 
eset = volt(lista,lista)
Egy új, eddig nem használt szekció következik, amely adatbázis keletkezését írja elõ. 

database 

volt(lista,lista)
Az adatbázis egy eleme egy szabály, annak is egy konstans behelyettesítési értékkel bíró esete, más néven egy tényállítás. 
Maga az adatbázis a számítógép RAM területén dinamikusan változhat, például az elsõ eleme elé vagy az utolsó eleme mögé új tény kerülhet, vagy onnan kitörlõdhet. A program futása közben az adatbázis nevének elõfordulásakor a tárban lévõ összes tényállítást figyelembe veszi a végrehajtó algoritmus, akár a program forrásszövegében szereplõ állításról volt szó, akár új -azt mondhatnánk : tanult- tényrõl. 
A database szekcióban tehát ennek az adatbázisnak a formális paramétereit kell megadnunk. Az adatbázis kezdeti tényállításait nem itt, hanem a clauses szekcióban, a szabályok és tények között kell leírni. 
Konkrét esetünkben ez arra jó, hogy két összetartozó változó értékét a volt szabály paramétereinek helyén tárolhassuk el. A szabályok formalizmusa már jórészt ismert elveken nyugszik, így nem foglalkozom velük, csak a szabályok részletes leírásánál utalok rá. 
Egyetlen kivétel -egyben újdonság- a következõ sor: 

ilyen(eset)
Különlegessége pedig az, hogy olyan paramétere van, amely a domains szekcióban elõírtak szerint nem az eddig ismert alap- és származtatott típusok egyike, hanem maga is egy szabálydeklaráció. 
Az most mellékes körülmény, hogy az eset típusként alkalmazott volt szabály adatbázisként tárolja a paraméterként megadott (lista,lista) formájú adatokat. Az ilyen formájú szabályok értelme például ott lehet, ahol a tárolt paraméterekre szükség van. A célkifejezést a program szövegébe foglaltam, hiszen csak egy megoldásra vagyok kíváncsi és pontosan ügyelni kell a kérdés formájára is. 
Most lássuk a szabályok sorát! A megoldás fõ algoritmusát már elemeztük, nézzük meg, miként realizálható ez Prologban. 

megoldBJ(Bvolt,Jvolt,Buj,Juj) if 
    viszBJ(Bvolt,Jvolt,Eza,Batm,Jatm) and 
    write("\n",Batm," >>",Eza,">> ",Jatm) and 
    megoldJB(Batm,Jatm,Juj,Buj). 
megoldJB([ures],_,_,_). 
megoldJB(Bvolt,Jvolt,Buj,Juj) if 
    viszJB(Bvolt,Jvolt,Eza,Batm,Jatm) and 
    write("\n",Batm," <<",Eza,"<< ",Jatm) and 
    megoldBJ(Batm,Jatm,Juj,Buj).
A megoldás szabálya tehát igazából két szabály, a B-J utazást jelentõ megoldBJ, és a J-B irányú mozgást leíró megoldJB. Itt lelhetõ fel a leírt algoritmus: 

· Elõször a megoldBJ rutint csinálja, ez átvisz egy utast, ezt kiírja, majd hívja a megoldJB rutint. 

· A megoldJB rutin két szabályból áll. Az elsõ elvégzi a vizsgálatot, és ha teljesül a végfeltétel, akkor leállítja a program munkáját. 

· A megoldJB második szabálya adja az algoritmus harmadik részét, vagyis egy J-B irányú utazást keres, azt kiírja és újra hívja az elõzõ, megoldBJ rutint, ezzel visszatér a program az egyetlen sikeres kilépést biztosító ciklushoz. 

Meg kell említenem, hogy az ilyen rekurzió-típust közvetett rekurziónak nevezik. 
A következõ lépés meghatározni, mit jelent a visz eljárás. Ez pedig nem más, mint kiválasztani egy utast a várakozók közül, áttenni a másik partra, majd megnézni, hogy az így keletkezett állapot megengedhetõ-e? 

viszBJ(Bvolt,Jvolt,Ezt,Buj,Juj) if 
    kivalaszt(Bvolt,Ezt) and 
    tesz(Ezt,Bvolt,Jvolt,Buj,Juj) and 
    enged(Buj,Juj). 
viszJB(Bvolt,Jvolt,Ezt,Buj,Juj) if 
    kivalaszt(Jvolt,Ezt) and 
    tesz(Ezt,Jvolt,Bvolt,Juj,Buj) and 
    enged(Buj,Juj).
Azért van két eljárás, mert a különbözõ irányú mozgások miatt más paraméterekhez kell hozzáadni illetve elvenni elemeket. 
Vegyük sorra az itt szereplõ három szabályt! 

kivalaszt( _ ,ures). 
kivalaszt(Ebbol,farkas) if bennevan(farkas ,Ebbol). 
kivalaszt(Ebbol,kaposzta) if bennevan(kaposzta ,Ebbol). 
kivalaszt(Ebbol,kecske) if bennevan(kecske ,Ebbol).
Tehát kiválaszt egyet az adott listából, ha benne van. 

tesz(ures,Innenvolt,Idevolt,Innenlett,[csonakos|Idevolt]) if 
    kivesz(Innenvolt,csonakos,Innenlett). 
tesz(Ezt,Innenvolt,Idevolt,Innenlett,[csonakos|[Ezt|Idevolt]]) if 
    Ezt<>"ures" and 
    kivesz(Innenvolt,csonakos,Innenatm) and 
    kivesz(Innenatm,Ezt,Innenlett).
Itt két eset van. Ha a kiválasztott a csónakos, akkor csak a csónakost kell az eredménylistához fûzni, ha másik, akkor azt és még a csónakost is. Mindezek azután történnek, hogy a máshova fûzött elemeket kivettük a helyükrõl, és meghatároztuk a maradékot. Ennek mûvelete a 

kivesz([E|L],E,L). 
kivesz([F|L1],E,[F|L2]) if kivesz(L1,E,L2).
rekurzióra épülõ szabálya. Az enged szabály a következõ: 

enged(Buj,Juj) if 
    not(ilyen(volt(Buj,Juj))) and 
    not(megeszi(Buj)) and 
    not(megeszi(Juj)) and 
    assertz(volt(Buj,Juj)).
Vagyis ha nem volt még olyan eset, mint ami keletkezett, és nem eszik meg senki semmit a bal- és a jobboldalon, akkor tárold az adatbázisban az állapotot, és ezzel az engedély meg van adva. 
Hogy hogyan döntsük el, hogy nem volt még ilyen eset? Itt az ilyen szabálya: 

ilyen(volt(Vbal,Vjobb)) if 
    volt(Bal,Jobb) and 
    permutacio(Vbal,Bal) and 
    permutacio(Vjobb,Jobb).
Ami nem mást fejez ki, minthogy volt már ilyen eset, ha a baloldali és a jobboldali névsort tartalmazó listák páronként egy volt adatbázisbeli eset két paraméterének valamelyik permutációi. 
Végül a megeszi szabályáról, amely rendkívül egyszerû: 

megeszi(L) if permutacio([farkas,kecske,ures],L). 
megeszi(L) if permutacio([kaposzta,kecske,ures],L). 
megeszi(L) if permutacio([farkas,kaposzta,kecske,ures],L).
Vagyis a listával adott szereplõk valamelyike megeszi a másikat, ha az a lista egy meghatározott lista valamely permutációja. Itt is három eset lehetséges, ezért van erre három szabály. 
A permutacio és a kiemel szabályok már ismertek, ezért nem is részletezem ezeket. Nos, íme a teljes forrásnyelvi szöveg: 

domains 

csonakos,farkas,kecske,kaposzta,ures = symbol 
lista = symbol* 
eset = volt(lista,lista)
database 

volt(lista,lista)
predicates 

megoldBJ(lista,lista,lista,lista) 
megoldJB(lista,lista,lista,lista) 
viszBJ(lista,lista,symbol,lista,lista) 
viszJB(lista,lista,symbol,lista,lista) 
tesz(symbol,lista,lista,lista,lista) 
kivalaszt (lista,symbol) 
enged(lista,lista) 
megeszi(lista) 
kivesz(lista,symbol,lista) 
bennevan(symbol,lista) 
ilyen(eset) 
permutacio(lista,lista) 
kiemel(symbol,lista,lista)
goal 

clearwindow and megoldBJ([csonakos,farkas,kecske,kaposzta,ures], 
         [ures], 
         [ures], 
         [csonakos,farkas,kecske,kaposzta,ures]).
clauses 

megoldBJ(Bvolt,Jvolt,Buj,Juj) if 
    viszBJ(Bvolt,Jvolt,Eza,Batm,Jatm) and 
    write("\n",Batm," >>",Eza,">> ",Jatm) and 
    megoldJB(Batm,Jatm,Juj,Buj). 
megoldJB([ures],_,_,_). 
megoldJB(Bvolt,Jvolt,Buj,Juj) if 
    viszJB(Bvolt,Jvolt,Eza,Batm,Jatm) and 
    write("\n",Batm," <<",Eza,"<< ",Jatm) and 
    megoldBJ(Batm,Jatm,Juj,Buj). 

viszBJ(Bvolt,Jvolt,Ezt,Buj,Juj) if 
    kivalaszt(Bvolt,Ezt) and 
    tesz(Ezt,Bvolt,Jvolt,Buj,Juj) and 
    enged(Buj,Juj). 
viszJB(Bvolt,Jvolt,Ezt,Buj,Juj) if 
    kivalaszt(Jvolt,Ezt) and 
    tesz(Ezt,Jvolt,Bvolt,Juj,Buj) and 
    enged(Buj,Juj). 

enged(Buj,Juj) if 
    not(ilyen(volt(Buj,Juj))) and 
    not(megeszi(Buj)) and 
    not(megeszi(Juj)) and 
    assertz(volt(Buj,Juj)). 

tesz(ures,Innenvolt,Idevolt,Innenlett,[csonakos|Idevolt]) if 
    kivesz(Innenvolt,csonakos,Innenlett). 
tesz(Ezt,Innenvolt,Idevolt,Innenlett,[csonakos|[Ezt|Idevolt]]) if 
    Ezt<>"ures" and 
    kivesz(Innenvolt,csonakos,Innenatm) and 
    kivesz(Innenatm,Ezt,Innenlett). 

ilyen(volt(Vbal,Vjobb)) if 
    volt(Bal,Jobb) and 
    permutacio(Vbal,Bal) and 
    permutacio(Vjobb,Jobb). 

kivalaszt( _ ,ures). 
kivalaszt(Ebbol,farkas) if bennevan(farkas ,Ebbol). 
kivalaszt(Ebbol,kaposzta) if bennevan(kaposzta ,Ebbol). 
kivalaszt(Ebbol,kecske) if bennevan(kecske ,Ebbol). 

megeszi(L) if permutacio([farkas,kecske,ures],L). 
megeszi(L) if permutacio([kaposzta,kecske,ures],L). 
megeszi(L) if permutacio([farkas,kaposzta,kecske,ures],L). 

kivesz([E|L],E,L). 
kivesz([F|L1],E,[F|L2]) if kivesz(L1,E,L2). 

bennevan(Ez,[Ez|_]). 
bennevan(Ez,[_|Maradek]) if bennevan(Ez,Maradek). 

permutacio(Xs,[Z|Zs]) if kiemel(Z,Xs,Ys) and permutacio(Ys,Zs). 
permutacio([],[]). kiemel(X,[X|Xs],Xs). 

kiemel(X,[Y|Ys],[Y|Zs]) if kiemel(X,Ys,Zs). 

volt(["csonakos","farkas","kaposzta","kecske","ures"],["ures"]). 
    /* kezdõ állapot */
A program futtatási képe a RUN után: 
  

	["farkas","kaposzta","ures"] >>kecske>> ["csonakos","kecske","ures"]   
["csonakos","farkas","kaposzta","ures"] <<ures<< ["kecske","ures"]  
["kaposzta","ures"] >>farkas>> ["csonakos","farkas","kecske","ures"]   
["csonakos","kecske","kaposzta","ures"] <<kecske<< ["farkas","ures"]  
["kecske","ures"] >>kaposzta>> ["csonakos","kaposzta","farkas","ures"]   
["csonakos","kecske","ures"] <<ures<< ["kaposzta","farkas","ures"]   
["ures"] >>kecske>> ["csonakos","kecske","kaposzta","farkas","ures"]


A listák között a kisebb és a nagyobb jelekkel határoltuk az utas nevét. Remélem, tanulságosnak bizonyult ez a feladat már csak a bevezetett új fogalmak és a közvetett rekurziónak nevezett programszerkezetnek köszönhetõen is. 

III.8. Repülõút tervezése

Egy utazási iroda számára repülõjáratok adatait tartalmazó információs táblát kell készítenünk. A táblán láthatóak az éppen elérhetõ járatok az adott városok között. A kiinduló és a célállomás adatainak bekérése után az összes lehetséges útvonalat meg kell adnunk. Adott a következõ sematikus térkép, ahol a vonalak a mindkét irányban használható repülõjáratokat jelentik: 

[image: image2]
Tényállításaink ezeket az útvonalakat tartalmazzák. 
A jarat szabály alkalmazásával megtakaríthatjuk azt, hogy a repülõjáratokat a visszaútjukkal is fel kelljen sorolnunk, hiszen a jarat akkor is igaz, ha elöl a kiinduló város és hátul a cél, illetve ha fordítva vannak a városok nevei. 
A goal szekció tartalmaz egy olyan beépített eljárást, amit eddíg nem alkalmaztunk. Ez az adatok beolvasására szolgáló readln eljárás, amelynek argumentuma az a stringváltozó, amelynek ezzel az eljárással akarunk értéket adni. 
Mivel az összes lehetséges útvonalra kíváncsiak vagyunk, a célkifejezés végére kell helyeznünk a fail szót, ami a kudarc mesterséges elõidézésére szolgál és arra készteti a rendszert, hogy kíséreljen meg a legutóbbi választási lehetõségtõl kiindulva egy másik megoldást találni. Mindaddíg ezt teszi, amíg van meg nem vizsgált eset. 
Az ut szabály négy paramétere közül az elsõ kettõ az aktuális start- és célállomást tartalmazzák, míg a harmadik paraméter azokat a városokat tartalmazza lista szerkezetben, amelyeket a keresés során érintett már a program. Ha a start és a cél megegyezik, akkor a szabály elsõ sora alapján a megoldás triviális. 
Ha az elsõ szabály nem teljesül, akkor lép érvénybe a második. Ez akkor teljesül, ha van olyan járat, amely a két város között közlekedik. 
Ha nincs ilyen járat, akkor a harmadik szabály igyekszik megoldást találni. Ez elõször keres egy tetszõleges járatot, megnézi, hogy az ezzel elérhetõ város szerepelt-e már az eddig bejárt útvonalban, és ha nem, akkor ebbõl a városból próbálja újból elérni a végcélt, miközben tárolja az útvonal-leírásban az új várost. 

domains 

varos = symbol 
varosok = varos* 
sorszam = integer
predicates 

ut(varos,varos,varosok,varosok) 
repulojarat(varos,varos) 
jarat(varos,varos) 
nemeleme(varos,varosok)
goal 

write("Melyik városból akarsz indulni ? ") and 
readln (Start) and 
write("Hová akarsz eljutni ? ") and 
readln (Cel) and 
ut(Start,Cel,[Start],Megoldas) and 
write(Megoldas) and 
nl and fail.
clauses 

ut(Varos,Varos,_,[Varos]). 
ut(Start,Cel,[_],[Start,Cel]) if 
    jarat(Start,Cel). 
ut(Start,Cel,Megtett,[Start|Ut]) if 
    jarat(Start,Uj_varos) and 
    nemeleme(Uj_varos,Megtett) and 
    ut(Uj_varos,Cel,[Uj_varos|Megtett],Ut). 

jarat(X,Y) if repulojarat(X,Y). 
jarat(X,Y) if repulojarat(Y,X). 

nemeleme(_,[]). 
nemeleme(Varos1,[Varos2|Varosok]) if 
    nemeleme(Varos1,Varosok) and 
    Varos1<>Varos2. 

repulojarat(budapest,london). 
repulojarat(moszkva,frankfurt). 
repulojarat(helsinki,moszkva). 
repulojarat(kiev,moszkva). 
repulojarat(london,lisszabon). 
repulojarat(frankfurt,helsinki). 
repulojarat(budapest,frankfurt). 
repulojarat(london,frankfurt).
A program néhány futtatási eredménye: 
  

	Melyik városból akarsz indulni ? lisszabon  
Hova akarsz eljutni ? kiev ["lisszabon","london","frankfurt","helsinki","moszkva","kiev"] ["lisszabon","london","frankfurt","moszkva","kiev"] ["lisszabon","london","budapest","frankfurt","helsinki","moszkva","kiev"] ["lisszabon","london","budapest","frankfurt","moszkva","kiev"]  

Melyik városból akarsz indulni ? budapest  
Hova akarsz eljutni ? london  
["budapest","london"]  
["budapest","frankfurt","london"]  

Melyik városból akarsz indulni ? frankfurt  
Hova akarsz eljutni ? frankfurt  
["frankfurt"]


  

III.9. Craig felügyelõ esetei

A Scotland Yard híres felügyelõjének sok esetét is említi Raymond Smullyan "Mi a címe ennek a könyvnek?" címû rendkívül érdekes könyve. A könyv számozása szerint most a 80. és 81. feladatot próbáljuk megoldani Prolog segítséggel. 
Az ügyben négy lehetséges vádlott van, és el kell dönteni, hogy bûnösek vagy ártatlanok, esetleg kétséges a szerepük az adott állítások alapján. Ezek az állítások az elsõ esetben a következõk: 

1. Ha A és B mindketten bûnösek, akkor C bûntárs. 

2. Ha A bûnös, akkor B és C közül legalább az egyik bûntárs. 

3. Ha C bûnös, akkor D bûntárs. 

4. Ha A ártatlan, akkor D bûnös. 

A döntést bízzuk a következõ programra! 

domains 

allitas = bunos ; artatlan 
itelet = allitas*
predicates 

kijelentes1(itelet) 
kijelentes2(itelet) 
kijelentes3(itelet) 
kijelentes4(itelet) 
valami(allitas) 
dontes(itelet)
clauses 

kijelentes1([bunos ,bunos ,bunos,D]) if valami(D). 
kijelentes1([artatlan,B ,C ,D]) if valami(B) and valami(C) and valami(D). 
kijelentes1([A ,artatlan,C ,D]) if valami(A) and valami(C) and valami(D). 

kijelentes2([bunos ,bunos,bunos,D]) if valami(D). 
kijelentes2([bunos ,bunos,C ,D]) if valami(C) and valami(D). 
kijelentes2([bunos ,B ,bunos,D]) if valami(B) and valami(D). 

kijelentes2([artatlan,B ,C ,D]) if valami(B) and valami(C) and valami(D). 
kijelentes3([A,B,bunos ,bunos]) if valami(A) and valami(B). 
kijelentes3([A,B,artatlan,D ]) if valami(A) and valami(B) and valami(D). 

kijelentes4([artatlan,B,C,bunos]) if valami(B) and valami(C). 
kijelentes4([bunos ,B,C,D ]) if valami(B) and valami(C) and valami(D). 

dontes(K) if 
    kijelentes1(K) and 
    kijelentes2(K) and 
    kijelentes3(K) and 
    kijelentes4(K). 

valami(artatlan). 
valami(bunos).
A program célja olyan feltevéseket generálni, amely a négy kijelentésnek egyaránt megfelelnek. Egy-egy ilyen kijelentés több szabállyal írható le, mivel ezek az implikációk több esetben is igazak lehetnek. 
A feltevések olyan listát alkotnak, aminek az elemei csak a bunos vagy az artatlan szavak lehetnek. A lista elsõ eleme az A-ra, a második a B-re stb. vonatkozik. 
Érdemes megvizsgálni az adott implikációk egyikének megfogalmazását, hiszen ehhez hasonló a többi is, és sok más logikai feladat is ehhez hasonló módon oldható meg. Válasszuk ki a második kijelentést, mivel ez a többinél valamivel bonyolultabb. 
  

Ha A bûnös, akkor B és C közül legalább az egyik bûntárs.
kijelentes2([bunos ,bunos,bunos,D]) if valami(D). 
kijelentes2([bunos ,bunos,C ,D]) if valami(C) and valami(D). 
kijelentes2([bunos ,B ,bunos,D]) if valami(B) and valami(D). 
kijelentes2([artatlan,B ,C ,D]) if valami(B) and valami(C) and valami(D).
Az elsõ sor azt az esetet veszi, amikor "A" bûnössége esetén "B" és "C" is bûnös. Ebben az esetben D valami tetszõleges (bûnös vagy ártatlan) érték. 
A második és a harmadik szabály szerint B és C közül csak az egyik a biztosan bûnös, és a többi valami tetszõleges. A negyedik szabály szerint ha A ártatlan, akkor a többi bármi lehet. A backtrack-et alkalmazó végrehajtó algoritmus az összes szabálykombinációt megvizsgálja, és kiírja a lehetséges eseteket. 
Lássuk ezt a futtatási képet! 
  

	Goal : dontes(M)  
M=[bunos,bunos,bunos,bunos]  
M=[bunos,bunos,bunos,bunos]  
M=[bunos,bunos,bunos,bunos]  
M=[artatlan,artatlan,artatlan,bunos]  
M=[artatlan,artatlan,bunos,bunos]  
M=[artatlan,bunos,artatlan,bunos]  
M=[artatlan,bunos,bunos,bunos]  
M=[artatlan,artatlan,artatlan,bunos]  
M=[artatlan,artatlan,bunos,bunos]  
M=[bunos,artatlan,bunos,bunos]  
10 Solutions 


Tehát az adott kijelentésekbõl az következik, hogy egyedül D-rõl tudhatjuk biztosan, hogy bûnös, a többi lehet ártatlan is, bûnös is. 
Craig felügyelõ másik esete szintén érdekes, és az elõzõ programhoz hasonlóan oldható meg Prologban. Csupán a kijelentések módosulnak a következõképpen: 

1. Ha A bûnös, akkor B bûntárs. 

2. Ha B bûnös, akkor vagy C bûntárs, vagy A ártatlan. 

3. Ha D ártatlan, akkor A bûnös, és C ártatlan. 

4. Ha D bûnös, akkor A is az. 

Az ennek megfelelõ program az alábbi: 

domains 

allitas = bunos ; artatlan 
itelet = allitas*
predicates 

kijelentes1(itelet) 
kijelentes2(itelet) 
kijelentes3(itelet) 
kijelentes4(itelet) 
valami(allitas) 
dontes(itelet)
clauses 

kijelentes1([bunos ,bunos,C,D]) if valami(C) and valami(D). 
kijelentes1([artatlan,B ,C,D]) if valami(B) and valami(C) and valami(D). 

kijelentes2([A ,bunos,bunos,D]) if valami(A) and valami(D). 
kijelentes2([artatlan,bunos,C ,D]) if valami(C) and valami(D). 

kijelentes3([bunos,B,artatlan,artatlan]) if valami(B). 
kijelentes3([A ,B,C ,bunos ]) if valami(A) and valami(B) and valami(C). 

kijelentes4([bunos,B,C,bunos ]) if valami(B) and valami(C). 
kijelentes4([A ,B,C,artatlan]) if valami(A) and valami(B) and valami(C). 

dontes(K) if 
    kijelentes1(K) and 
    kijelentes2(K) and 
    kijelentes3(K) and 
    kijelentes4(K). 

valami(bunos). 
valami(artatlan).
A program mûködése az elõzõhöz teljesen hasonló, most csak a végeredményt mutatom, ahogyan a képernyõn látható: 
  

	Goal : dontes(M)  
M=[bunos,bunos,bunos,bunos]  
1 Solution


Vagyis mind a négyen bûnösek a kijelentések alapján. A fenti két program végeredménye összhangban van a könyv következtetéseivel, így ha az helyes, akkor a programunk is helyesen mûködik (implikáció !). 

III.10. N-királynõ a sakktáblán

A feladat iskolapéldája a backtrack algoritmusnak. A Prolog rendszerben eleve benne van ez az algoritmus, csak a megfelelõ környezetet kell hozzá biztosítani. 
Feladatunk tehát az, hogy N darab királynõt (más elnevezéssel vezért) kell az NxN-es sakktáblán elhelyezni úgy, hogy ne üssék egymást, azaz ne essenek egy sorba, egy oszlopba vagy azonos átlóba. 
Az alapötlet az, hogy a királynõk elhelyezkedését az [1,...,8] intervallum egy permutációja képviselje, ahol a pozíció jelenti a sorokat és az érték azt, hogy azon a soron belül hányadik oszlopban van a sakkfigura. 
A program fõeljárása a következõ: 

kiralyno(N,Kl) if sorozat(1,N,Ns) and permutacio(Ns,Kl) and johely(Kl) and nl.
Ez tehát három részfeladatra bontotta a problémát. Az elsõ teendõ az, hogy elõállítsunk egy sorozatot 1-tõl N-ig az Ns listában. Ennek megoldása elég egyszerû: 

sorozat(M,N,[M|Ns]) if M<N and M1=M+1 and sorozat(M1,N,Ns). 
sorozat(N,N,[N]).
Vagyis amíg a végértékhez el nem értünk, addíg csökkentjük a feladat tartományát azáltal, hogy a kezdõértéket eggyel nagyobbra tesszük. Ha elérte, akkor megvan a lista utolsó eleme. Ezek után a rekurziók sora visszafordul, és sorra fûzi az elejéhez a következõ, eggyel kisebb számot. 
A második részfeladat is ismert, hiszen az elõállított Ns listát kell permutálnia. Errõl már sokat beszéltem, most ne is errõl, hanem a permutáció eljárás alkalmazásából következõ dolgokról szólok. Az ugyanis nyilvánvaló, hogy így sok olyan permutációt elõáll, amely rossz megoldásnak bizonyul. Ha nem a vaktában próbálkozást programozzuk, akkor olyan algoritmust kell alkalmaznunk, amely a már elhelyezett királynõk helyzetét figyelembe véve választ a következõnek új helyet. Mi azonban ne legyünk még ilyen okosak. Ha már megvan egy permutáció, akkor utólag döntsük el, hogy jó-e ez nekünk vagy sem. Erre szolgáló függvény a johely. 

johely([K|Kl]) if johely(Kl), not(leut(K,Kl)). 
johely([]).
Tehát a lista jó helyeket jelent, ha a Kl már jó és ehhez úgy illeszkedik a K érték, hogy nem üti le a többi figurát. Most tehát a leut függvényt kell megoldanunk. 

leut(X,Xs) if leut1(X,1,Xs). 
leut1(X,N,[Y|Ys]) if X=Y+N or X=Y-N. 
leut1(X,N,[_|Ys]) if N1=N+1 and leut1(X,N1,Ys).
Láthatóan egy iteratív szabállyal állunk szemben, amely egy rekurzív szabályt készít elõ és hív meg. Az utóbbi az érdekesebb. Az elsõ sora azt nézi meg, hogy az alatta N-el lévõ sorban az átlók irányában van-e már királynõ. Ha nincs, akkor érvénybe lép a második szabály, amely szerint meg kell vizsgálni az eggyel lejjebbi soron lévõ királynõt. 
Mivel elegendõ az elõzõ vizsgálatok miatt csak lefelé vizsgálni, ezért elegendõ az elõzõ figura nélküli listára végezni a vizsgálatot. Nincs más teendõnk, mint a teljes programlista és néhány futtatási példa áttekintése: 

domains 

szam = integer 
lista= szam*
predicates 

kiralyno(szam,lista) 
sorozat(szam,szam,lista) 
johely(lista) 
leut(szam,lista) 
leut1(szam,szam,lista) 
permutacio(lista,lista) 
kiemel(szam,lista,lista)
clauses 

kiralyno(N,Kl) if sorozat(1,N,Ns) and permutacio(Ns,Kl) and johely(Kl). 

johely([K|Kl]) if johely(Kl), not(leut(K,Kl)). 
johely([]). 

leut(X,Xs) if leut1(X,1,Xs). 
leut1(X,N,[Y|Ys]) if X=Y+N or X=Y-N. 
leut1(X,N,[_|Ys]) if N1=N+1 and leut1(X,N1,Ys). 

permutacio(Xs,[Z|Zs]) if kiemel(Z,Xs,Ys),permutacio(Ys,Zs). 
permutacio([],[]). 

kiemel(X,[X|Xs],Xs). 
kiemel(X,[Y|Ys],[Y|Zs]) if kiemel(X,Ys,Zs). 

sorozat(M,N,[M|Ns]) if M<N and M1=M+1 and sorozat(M1,N,Ns). 
sorozat(N,N,[N]).
Íme néhány eredmény és egy-két megoldás szemléltetése: 
  

	Goal: kiralyno(6,K)  
K=[2,4,6,1,3,5]  
K=[3,6,2,5,1,4]  
K=[4,1,5,2,6,3]  
K=[5,3,1,6,4,2]  
  4 Solutions  
Goal: kiralyno(5,K)  
K=[1,3,5,2,4]  
K=[1,4,2,5,3]  
K=[2,4,1,3,5]  
K=[2,5,3,1,4]  
K=[3,1,4,2,5]  
K=[3,5,2,4,1]  
K=[4,1,3,5,2]  
K=[4,2,5,3,1]  
K=[5,2,4,1,3]  
K=[5,3,1,4,2]  

10 Solutions
	K=[2,4,6,1,3,5]
.$....
...$..
.....$
$.....
..$...
....$.
K=[4,1,5,2,6,3]
...$..
$.....
....$.
.$....
.....$
..$...
	K=[1,3,5,2,4]
$....
..$..
....$
.$...
...$.
K=[1,4,2,5,3]
$....
...$.
.$...
....$
..$..


Láthatóan sok olyan megoldás született, amelyek egymás tükörképei csupán. Így a sakktáblák szimmetriatengelyeinek számai is befolyásolják a megoldások számait. Felsorolok néhány megoldás-számot különbözõ N értékekre: 
  

	N=3
	0

	N=4
	2

	N=5
	10

	N=6
	4

	N=7
	40

	N=8
	92

	N=9
	352


  

Zárszó

Befejezni egy ilyen munkát szerintem nem lehet tisztességgel. Azért nem, mert olyan sok feladattípus maradt ki, hogy óhatatlanul hiányérzetet kelt. Ezt belátva arra buzdítom a tisztelt Olvasót, fogjon hozzá újabb feladatok megoldásához Prolog nyelven, és tegye közkinccsé munkája eredményét. De ha nem is teszi ezt, akkor is az elme gyémántkeménységû csiszolóeszköze marad a Prolog, a jövõ egyik lehetséges programozási nyelve. Végül mindenki szíves figyelmébe ajánlom a Függeléket, amely a Turbo Prolog beépített függvényeit, eljárásait szótárszerûen, tömör definíciókkal kiegészítve sorolja fel. További jó munkát, a munkában sok örömet !
Függelék

Az alábbi összefoglalást a Turbo Prolog Felhasználói Kézikönyvbõl fordítottam. Elnézést kérek a néhol nem túl szabatos megfogalmazásokért, de remélem, a Turbo Prolog jelenlegi és majdani használói egyaránt hasznosnak ítélik igyekezetemet. 

Aritmetikai függvények és predikátumok

  

	bitand(X,Y,Z)
	Ha X és Y egész számok, akkor Z értéke X-en és  Y-on mint 16 bites számokon végzett logikai ÉS,  VAGY, NEM és KIZÁRÓ VAGY mûveletek után kapott  értéke

	bitor(X,Y,Z)
	

	bitnot(X,Z)
	

	bitxor(X,Y,Z)
	

	bitleft(X,N,Y)
	Ha X és N egész értékek, akkor Y értéke egyenlõ  a 16 bites számnak tekintett X bitjeinek N bittel balra illetve jobbra tolásával kapott értékkel

	bitright(X,N,Y)
	

	random(X)
	pszeudo-véletlen szám (0<=X<1 )

	X mod Y
	X-nek Y-al való osztási maradéka

	X div Y
	X-nek Y-al való egészosztásának eredménye

	abs(X)
	X abszolút értéke

	cos(X)
	X radián trigonometrikus függvényértéke

	sin(X)
	

	tan(X)
	

	arctan(X)
	X valós számhoz tartozó arkusztangens érték

	exp(X)
	e az X-edik hatványon

	ln(X)
	X-nek e alapú logaritmusa

	log(X)
	X-nek 10 alapú logaritmusa

	sqrt(X)
	X négyzetgyöke


  
  

A színek attribútumai

  

	Háttér színek
	Elõtér színek

	0
	fekete
	0
	fekete

	8
	szürke
	1
	kék

	16
	kék
	2
	zöld

	24
	világoskék
	3
	cián

	32
	zöld
	4
	piros

	40
	világoszöld
	5
	bíbor

	48
	cián
	6
	barna

	64
	piros
	7
	fehér

	72
	világos piros
	A két értéket össze kell adni
Villogtatás: +128

	80
	bíbor
	

	88
	világos bíbor
	

	96
	barna
	

	104
	sárga
	

	112
	fehér
	

	120
	intenzív fehér
	


  

A standard (beépített) eljárások alfabetikus leírása

Minden eljárást a következõ formátumban írjuk le: 

· az eljárás neve és tipikus hívási módja 

· az eljárásparaméterek típusai a megfelelõ sorrendben 

· a lehetséges változóértékek listája 

· az adott változóérték melletti hatás leírása 

  

	asserta(<kijelentés>)
	(dbasedom):(i)

	(i)
	Beilleszti a kijelentést a RAM adatbázisba az azonos nevû megfelelõ kijelentések elé. A kijelentésnek a programon belül adatbázisnak deklarált kifejezésnek kell lennie.


  

	assertz(<kijelentés>)
	(dbasedom):(i)

	(i)
	Beilleszti a kijelentést a RAM adatbázisba az azonos nevû megfelelõ kijelentések mögé. A kijelentésnek a programon belül adatbázisnak deklarált kifejezésnek kell lennie.


  

	attribute(Attr)
	(integer):(i)(o)

	(i) 
	Beállítja a default attribútum értékét minden képernyõpozícióra

	(o)
	Az aktuális default attribútumot állítja Attr értékre, minden képernyõpozícióra


  

	back(Step)
	(integer):(i)

	(i)
	Hatására a teknõc az aktuális pozíciójától az aktuális irányában Step lépést lép hátrafelé. Kudarccal tér vissza, ha az elérni kívánt pozíció a képernyõn kívülre esik. A képernyõ 32000 függõleges és 32000 vízszintes pozícióból áll. A teknõc aktuális pozíciója csak akkor változik, ha a back teljesülhet.


  

	beep
	-

	(i)
	Megszólaltatja a számítógép hangszóróját.


  

	bios(InterruptNo,RegsIn,RegsOut)
	(integer,regdom,regdom):(i,i,o)

	(i)
	InterruptNo számmal jelzett megszakítást generál a RegsIn regiszterértékekkel és a megszakításból való visszatérés után a RegsOut tartalmazza a regiszter értékét.A RegsIn és a RegsOut belsõ típusok az alábbi módon vannak definiálva.  
regdom = reg(AX,BX,CX,DX,SI,DI,DS,ES) ahol AX,BX,CX,DX,SI,DI,DS,ES mindegyike egész típusú.


  

	bound(Variable)
	(<Variable>):(o)

	(i)
	Teljesül, ha a változó meghatározott.


  

	char_int(CharParam,IntParam)
	(char,integer):(i,o)(o,i)(i,i)

	(i,o)
	CharParam ASCII kódját rendeli IntParam-hoz

	(o,i)
	A decimális IntParam által meghatározott ASCII jelet rendeli CharParam-hoz

	(i,i)
	Teljesül, ha az IntParam által meghatározott ASCII jel megegyezik a CharParam jelével


  

	clearwindow
	-

	(i)
	Törli az éppen aktív szöveges ablakot és feltölti a hozzárendelt alapszínnel


  

	closefile(SymbolicFileName)
	(file):(i)

	(i)
	Lezárja a megnevezett file-t. Akkor válik sikeressé, ha a file már nem nyitott


  

	consult(DOSFileName)
	(string):(i)

	(i)
	Egy szövegfile-t olvas az aktuális adatbázishoz (ez a szövegfile lehet például egy save-el kimentett adatbázis tartalma is.). A predikátum sikeres, ha a memóriába töltötte az összes file-beli tényállítást (amelyek az adatbázis deklarációjának megfelelnek). Ha a szövegfile bármiféle szintaktikai hibát tartalmaz, a consult sikertelen lesz.


  

	cursor(Row,Column)
	(integer,integer):(i,i)(o,o)

	(i,i)
	Az aktuális szöveges ablakban a bal felsõ (0,0) pozícióhoz viszonyítva áthelyezi a kurzor pozícióját.

	(o,o)
	Row és Column a kurzor pozícióinak megfelelõ értékeket kapják.


  

	cursorform(Startline,Endline)
	(integer,integer):(i,i)

	(i)
	A kurzor vastagságát és f-ggõleges pozícióját állítja be egy karakterpozíción belül. Minden karakter 14 sorból áll, így a két paraméter 1 és 14 közötti számokat vehet fel, a szélsõértékeket is beleértve.


  

	date(Year,Month,Day)
	(integer,integer,integer):(i,i,i)(o,o,o)

	(i,i,i)
	Beolvassa a dátumot a számítógép belsõ órájáról.

	(o,o,o)
	Beállítja a dátumot a belsõ órán.


  

	deletefile(DosFileName) 
	(string):(i)

	(i)
	Törli a nevezett file-t az aktuális meghajtón.


  

	dir(PathName,FileSpecString,DosFileName)
	(string,string,string)(i,i,o)

	(i)
	A Turbo Prolog directory-parancsát hívja. A jelölt útvonal és file-specifikáció meghatározza azokat a file-neveket, amelyek megjelennek a directory-ablakban. A felhasználó választhat a file-nevek között a kurzormozgató billentyûkkel, és ha ENTER-t nyomhat,ha a kívánt filenév lesz kiemelve, ekkor a file-névvel tér vissza a parancs. Kudarccal lép vissza, ha az ESC billentyût nyomták le.


  

	disk(DosPath)
	(string):(i)(o)

	(i)
	Beállítja a default meghajtót és útvonalat.

	(o)
	Visszatér az aktuális meghajtóval és útvonallal.


  

	display(String)
	(string):(i)

	(i)
	Megjeleníti a string tartalmát az aktuális ablakban. A tartalom vizsgálható (de meg nem változtatható) az editor vezérlõ billentyûivel.


  

	dot(Row,Column,Color)
	(integer,integer,integer):(i,i,i)(i,i,o)

	(i,i,i)
	Feltéve, hogy a képernyõ elõzetesen grafikus üzemmódban van, egy pontot rajzol a sor és oszlop koordinátákban, a kívánt színben. A koordináták 0 és 31999 közötti egész számok lehetnek, függetlenül a képernyõ üzemmódjától.

	(i,i,o)
	Ha a Color szabad változó, akkor a kérdéses pont színértékével tér vissza.


  

	edit(InputString,OutputString)
	(string,string):(i,o)

	(i)
	A Turbo Prolog editorát hívja be. Az InputString nevû file szerkeszthetõ az aktuális ablakban, kialakítva az OutputString-et.


  

	editmsg(InStr,OutStr,LeftHeader,RightHeader,  
          Message,HelpFileName,Position,Code)
	(string,string,string,string,string,string,string,integer,integer):  
(i,o,i,i,i,i,i,o)

	(i)
	A Turbo Prolog editorát hívja. Az InpStr nevû file szerkeszthetõ az aktuális ablakban, kialakítva az OutStr-et. A LeftHeader-el és a RighHeader-el adott két szöveg szúrható be a fejlécbe. A kurzor az Instr Position-odik helyére ugrik, és a jelzett Message az ablak alján látható. Az F1 lenyomásakor betöltendõ file neve HelpFileName. A Code azzal az értékkel tér vissza, ahogyan a szerkesztés véget ért (Code=0, ha az F10-el, Code=1 ha az ESC billentyûvel szakadt meg).


  

	eof(SymbolicFileName)
	(file):(i)

	(i)
	Megvizsgálja, hogy az aktuális file mutatója elérte-e a file végét (lásd filepos). Ha igen, akkor az eof teljesül, különben nem.


  

	existfile(DosFileName)
	(string):(i)

	(i)
	Teljesül, ha a filenév jelen van a meghajtó aktuális könyvtárában exit Megállítja a program futását és visszaadja a vezérlést a Turbo Prolog menürendszerének. fail Kudarcot vált ki a kijelentésre, így backtrack-et okozhat.


  

	field_attr(Row,Column,Lenght,Attr)
	(integer,integer, integer,integer):(i,i,i,i)(i,i,i,o)

	(i,i,i,i)
	Ha a Row és az Column egy képernyõpozíciót határoz meg az aktuális ablakban (lásd makewindow és shiftwindow), és ettõl a pozíciótól kezdve a Lenght hossz az ablakon belülre esik, akkor ezekre a képernyõpozíciókra beállítja az Attr értéket.

	(i,i,i,o)
	Az aktuális ablak Row-al és Column-al meghatározott Lenght hosszúságú helyén lévõ karakterek attribútuma értékét kapja meg az Attr változó. Ahogyan az elõbb, most is az ablakon bel-lre kell esnie a kijelölt mezõnek.


  

	field_str(Row,Column,Length,String)
	(integer,integer, integer,integer):(i,i,i,i)(i,i,i,o)

	(i,i,i,i)
	Ha a Row és az Column egy képernyõpozíciót határoz meg az aktuális ablakban (lásd makewindow és shiftwindow), és ettõl a pozíciótól kezdve a Lenght hossz az ablakon belülre esik, akkor ezekre a képernyõpozíciókra kiírja az String értékét az alábbi feltételek szerint(ha a String hossza nagyobb, mint a Length által adott érték, akkor az elsõ Length karakter lesz kiírva. Ha a String rövidebb, mint a Length, akkor a mezõ maradéka szóközzel lesz kitöltve.

	(i,i,i,o)
	Az aktuális ablak adott pozíciójában kezdõdõ Lenght hosszúságú karaktersorozatot olvassa be a String változóba. Ahogyan az elõbb, most is az ablakon belülre kell esnie a kijelölt mezõnek.


  

	filepos(SymbolicFileName,FilePosition,Mode)
	(file,real,integer):(i,i,i)(i,o,i)

	(i,i,i)
	A nevezett file-ban kiválasztja azt a pozíciót, ahova egy write utasítás adatot írjon. A pozíció számításakor figyelembe kell venni a file-ban rögzítendõ elemek típusaitól és a Mode értékétõl. Igy ha a SymbolicFileName egy byte-okból álló file-t jelent, és a FilePosition és a Mode értéke rendre 11 és 0, a file-ba írandó következõ byte a 11. pozícióra ker-l (a file elejétõl számítva)

	(i,o,i)
	Azzal a pozícióval tér vissza, ahova a következõ write utasítás helyezné az adatot. A file-pozíció értékét a Mode=0 érték mellett adja meg.

	Mód

Pozíció

0

A file elejéhez viszonyítva

1

A pillanatnyi pozícióhoz viszonyítva

2

A file végéhez viszonyítva

 


  

	file_str(DosFileName,StringVariable)
	(string,string):(i,o)

	(i)
	A nevezett file-ból beolvas (maximum 64K) karaktert a sztringváltozóba, amíg file-vége jelet nem talál (ASCII kódja 26).


  

	findall(Variable,<atom>,ListVariable)
	-

	(i)
	Listába írja az értékeket a backtrack végrehajtásából. Igy ha egy <atom> érvényes változókkal rendelkezõ kijelentés és a Variable egy változónév az argumentumai köz-l, akkor a ListVariable értéke ennek a változónak a backtrack során a szabály teljesülésekor felvett helyettesítési értékei által lesz meghatározva.


  

	flush(SymbolicFileName)
	(file):(i)

	(i)
	A belsõ file-puffer tartalmát kiviszi az aktuális írásra megnyitott egységre. A flush akkor használatos, ha az output közvetlenül a soros portra van irányítva és sz-kségessé válik az adatok kiküldése, mielõtt a puffer megtelik. (Normális esetben a puffer tartalma automatikusan kerül a kiviteli eszközre.)


  

	forward(Step)
	(integer):(i)

	(i)
	Feltéve, hogy a képernyõ grafikus módban van, a forward a tollat az aktuális pozíciótól Step lépéssel az aktuális irányban elõremozgatja. Kudarccal tér vissza, ha a mozgás elhagyja a képernyõ területét (a képernyõ 32000 vízszintes és függõleges lépésbõl áll). A teknõc aktuális pozíciója csak a forward sikeres befejezése után kap új értéket. Ha a toll le van téve, akkor vonalat húz maga után az érvényes tollszínnel.


  

	free(Variable)
	(<Variable>):(i)

	(i)
	Teljesül, ha a változó határozatlan.


  

	frontchar(String,FrontChar,RestString)
	(string,char,string):(i,o,o)(i,i,o)(i,o,i)(i,i,i)(o,i,i)

	(i)
	Mûködése az alábbi egyenlõség alapján történik: String=(a FrontChar és a RestString konkatenációja) tehát vagy a String-nek, vagy a FrontChar és a RestString-nek határozott értékûnek kell lennie.


  

	frontstr(NumberOfChars,String1,StarStr,String2)
	(integer,string,string,string):(i,i,o,o)

	(i)
	String1-et két részre vágja. A StartStr az elsõ NumberOfChars karaktert tartalmazza, az String2 pedig a maradékot.


  

	fronttoken(String,Token,Reststring)
	(string,string,string):(i,o,o)(i,i,o)(i,o,i)(i,i,i)(o,i,i)

	(i)
	Az alábbi egyenlõség alapján mûködik( String = (a Token és a RestString konkatenációja) vagyis vagy a String van meghatározva, vagy a Token és a RestString határozott. Igy a fronttoken teljes-l, ha a Token a String elsõ karaktereként és a RestString a String maradékaként van meghatározva. Egy vagy több karakter tokent képez az alábbi esetekben:  

  egy Turbo Prolog szintaxisnak megfelelõ <nev>-et alkotnak  

  egy Turbo Prologban érvényes valós vagy egész számot leíró karaktersorozatot alkotnak (az elõjel különálló tokenként tér vissza)  

  egyetlen, szóköztõl (ASCII 32) különbözõ karakter  


  

	graphics(ModeParam,Palette,Background) (integer,integer,integer):(i,i,i)
	-

	(i)
	Inicializálja a képernyõt közepes, nagy illetve extra-nagy felbontású grafikusra. ModeParam értéke állítja be a felbontást

	 

ModeParam
Oszlop
Sor
Adapter
Felbontás, színek
1

320

200

CGA

közepes, 4 szín

2

640

200

CGA

nagy, fekete-fehér

3

320

200

EGA

közepes, 16 szín

4

640

200

EGA

nagy, 16 szín 

5

640

350

EGA

3 szín

Background értéke (szintén egész érték) választja meg a háttér színét.


  

	inkey(Variable)
	(char):(i)

	(i)
	A lenyomott billentyû jelével tér vissza. Kudarc, ha nem volt lenyomva billentyû.


  

	isname(StringParam)
	(string):(i)

	(i)
	Teljes-l, ha a StringParam egy Turbo Prolog szintaxisának megfelelõ <nev>.


  

	left(Angle)
	(integer):(i)(o)

	(i)
	Elfordítja a teknõcöt Angle fokkal balra.

	(o)
	A teknõc irányát (fokokban) elhelyezi az Angle változóba.


  

	line(Row1,Col1,Row2,Col2,Color)
	(integer,integer,integer,integer,integer):(i,i,i,i)

	(i)
	Feltéve, hogy a képernyõ grafikus üzemmódban van, a line egy Color színû szakaszt húz a paraméterek által megjelölt pontok közé. A koordináták a [0..31999] intervallumba esnek, függetlenül a képernyõ üzemmódjától.


  

	makewindow(WindowNo,ScrAttr,FrameAttr,  
Header,Row,Col,Height,Width)
	(integer,integer,integer,string,integer,integer,integer,integer) :(i,i,i,i,i,i,i,i)

	(i)
	A képernyõ egy ter-letét ablaknak definiálja. Minden ablaknak van megk-lönböztetõ száma (WindowNo), amellyel lehet választani az ablakok között.Ha FrameAttr értéke nem 0, akkor keretet rajzol a jelölt ter-let köré, és a felsõ vonalba foglalja a Header szövegét. Amikor az ablakot definiáltuk, akkor az ablak törlõdik és a kurzor az ablak bal felsõ sarkába ugrik. A Row és Col értékei az ablak bal felsõ sarkának pozíciói a teljes képernyõter-leten, míg a Height és a Width az ablak magassága és szélessége. Fontos, hogy ezek a paraméterek összhangban legyenek a képernyõmérettel.


  

	membyte(Segment,Offset,Byte)
	(integer,integer,integer) :(i,i,i)(i,i,o)

	(i,i,i)
	Ha a Byte értéke ismert, akkor a Byte értékét a memóriának a Segment-el és Offset-el meghatározott címére teszi.

	(i,i,o)
	Ha a Byte értéke nem ismert, akkor a megadott tárcímrõl vett értéket teszi a Byte változóba.

	A tárcím számítása : tárcím=Segment*16+Offset


  

	memword(Segment,Offset,Word)
	(integer,integer,integer) :(i,i,i)(i,i,o)

	(i,i,i)
	Ha a Word értéke ismert, akkor a Word tartalmát a memóriának a Segment-el és Offset-el meghatározott címére teszi.

	(i,i,o)
	Ha a Word értéke nem ismert, akkor a megadott tárcímrõl vett értéket teszi a Word változóba.

	A tárcím számítása : tárcím=Segment*16+Offset 

A 32768 és 65536 közötti számokat negatív egészként értelmezi


  

	nl
	-

	(i)
	Kocsi-vissza és soremelés jelszekvenciát küld az aktív kimeneti egységre.


  

	not(<atom>)
	-

	(i)
	Teljesül, ha <atom> olyan célkifejezés, ami a végrehajtáskor kudarccal végzõdik.


  

	openappend(SymbolicFileName,DosFileName)
	(file,string):(i,i)

	(i)
	 Egy DosFileName nevû DOS-file-t nyit meg hozzáírásra, és a SymbolicFileName nevet rendeli hozzá a Turbo Prolog programbeli további hivatkozások céljára.


  

	openmodify(SymbolicFileName,DosFileName)
	(file,string):(i,i)

	(i)
	Egy DosFileName nevû DOS-file-t nyit meg írásra és olvasásra, és a SymbolicFileName nevet rendeli hozzá a Turbo Prolog programbeli további hivatkozások céljára. Ezt a predikátumot a filepos predikátummal együtt használhatjuk a véletlenelérésû adatfile-ok adatkezelésére.


  

	openread(SymbolicFileName,DosFileName)
	(file,string):(i,i)

	(i)
	Egy DosFileName nevû DOS-file-t nyit meg olvasásra, és a SymbolicFileName nevet rendeli hozzá a Turbo Prolog programbeli további hivatkozások céljára.


  

	openwrite(SymbolicFileName,DosFileName)
	(file,string):(i,i)

	(i)
	Egy DosFileName nevû DOS-file-t nyit meg írásra, és a SymbolicFileName nevet rendeli hozzá a Turbo Prolog programbeli további hivatkozások céljára. Ha DosFileName nevû file már létezik, akkor az törölve lesz.


  

	pencolor(Color)
	(integer):(i)

	(i)
	Meghatározza a toll által húzott vonal színét. 


  

	pendown
	-

	(i)
	Leteszi a tollat a forward és a back predikátumok számára.


  

	penup
	-

	(i)
	Felemeli a tollat a forward és a back predikátumok számára.


  

	portbyte(PortNo,Value)
	(integer,integer):(i,i)(i,o)

	(i,o)
	A PortNo számú I-O porton lévõ értéket teszi a Value változóba

	(i,i)
	A Value értékét küldi a PortNo számú I-O portra.


  

	ptr_dword(StringVar,Segment,Offset)
	(string,integer,integer): (i,o,o)(o,i,i)

	(i,o,o)
	Ha a StringVar ismert, akkor a StringVar címének Segment és Offset értékével tér vissza

	(o,i,i)
	Ha a Segment és az Offset ismert, akkor a címtõl kezdõdõ stringként értelmezhetõ stringgel tér vissza. A string ASCII kódokként értelmezett számokat tartalmaz az elsõ NUL byte-ig (aminek a kódja 0).


  

	readchar(CharVariable)
	(char):(o)

	(i)
	Beolvas egyetlen karaktert az aktuális bemeneti egységrõl (ami a billentyûzet, hacsak meg nem változtattuk azt a readdevice utasítással).


  

	readdevice(SymbolicFileName)
	(symbol):(i)(o)

	(i)
	Átállítja az aktuális bemeneti egységet a SymbolicFileName nevû megnyitott file-ra, amely vagy az elõre definiált keyboard vagy bármely más, olvasásra vagy módosításra megnyitott felhasználói file szimbolikus neve lehet.

	(o)
	A SymbolicFileName változóba teszi az éppen olvasásra megnyitott egység szimbolikus nevét, amely lehet keyboard vagy bármely más, olvasásra vagy módosításra megnyitott felhasználói file szimbolikus neve lehet. 


  

	readint(IntVariable)
	(integer):(o)

	(i)
	Beolvas egy egész számot az aktuális bemeneti egységrõl (ami a billentyûzet, hacsak meg nem változtattuk azt a readdevice utasítással). A bevitel végét a Carriage-Return jel jelenti.


  

	readln(StringVariable)
	(string):(o)

	(i)
	Karaktereket olvas be az aktuális bemeneti egységrõl (ami a billentyûzet, hacsak meg nem változtattuk azt a readdevice utasítással). A bevitel végét a Carriage-Return jel jelenti.


  

	readreal(RealVariable)
	(real):(o)

	(i)
	Beolvas egy valós számot az aktuális bemeneti egységrõl (ami a billentyûzet, hacsak meg nem változtattuk azt a readdevice utasítással). A bevitel végét a Carriage-Return jel jelenti.


  

	readterm(Domain,Term)
	(<name>,<variable>):(o,i)

	(i)
	Beolvas bármilyen szöveget, amit a write utasítás írt ki. Term a szöveg olvasásával meghatározott, feltéve, hogy összefér a Domain deklarációjával. A readterm lehetõvé teszi szabályok hozzáfûzését egy file-hoz.


  

	removewindow
	-

	(i)
	Eltávolítja az aktuális ablakot a képernyõrõl.


  

	renamefile(OldDosFileName,NewDosFileName)
	(string,string):(i,i)

	(i)
	Átnevezi az éppen elérhetõ lemezen az OldDosFileName nevû file-t NewDosFileName nevûre.


  

	retract (<fact>)
	(dbasedom):(i)

	(i)
	Törli az elsõ tényállítást az adatbázisból, amelyik illeszkedik az adott <fact> állítással.


  

	right(Angle)
	(integer):(i)(o)

	(i)
	Elfordítja a teknõcöt Angle fokkal jobbra.

	(o)
	A teknõc irányát (fokokban) elhelyezi az Angle változóba.


  

	save(DosFileName)
	(string):(i)

	(i)
	Elmenti az adatbázis összes szabályát a DosFileName nevû szöveges file-ba. Minden szabályt egy külön sorba ír. A file késõbb a consult predikátummal beolvasható. A szöveges file - így az egész adatbázis késõbb az editorral kezelhetõ.


  

	scr_attr(Row,Col,Attr)
	(integer,integer,integer):(i,i,i)(i,i,o)

	(i,i,i)
	A Row,Col pozícióban lévõ karakter attribútumát állítja be Attr értékre.

	(i,i,o)
	A Row,Col pozícióban lévõ karakter attribútumát adja meg az Attr változóban.


  

	scr_char(Row,Column,Char)
	(integer,integer,char):(i,i,i)(i,i,o)

	(i,i,i)
	A Char karaktert írja a képernyõ adott pozíciójára az érvényes attribútummal.

	(i,i,o)
	Beolvas egy karaktert az adott pozícióról.


  

	shiftwindow(WindowNo)
	(integer):(i)(o)

	(i)
	Átváltja az aktuális ablakot a WindowNo számúra. Minden más ablak a pillanatnyi állapotában marad. A kurzorpozíció visszatér oda, ahol akkor volt, amikor utoljára aktív volt ez az ablak.

	(o)
	WindowNo az aktív ablak számát kapja.


  

	sound(Duration,Freqency)
	(integer,integer):(i,i)

	(i)
	Hangot bocsát ki az adott frekvencián Duration századmásodpercig.


  

	storage(StackSize,HeapSize,TraiSize)
	(real,real,real):(o,o,o)

	(i)
	A Turbo Prolog rendszer által használt három run-time memória nagyságával tér vissza.


  

	str_char(StringParam,CharParam)
	(string,char):(i,o)(o,i)(i,i)

	(i,o)
	CharParam értékének azt az egyetlen karaktert teszi, amelyet az StringParam határoz meg.

	(o,i)
	StringParam értéke a CharParam által meghatározott karakter lesz.

	(i,i)
	Teljesül, ha mindkét paraméter ugyanazt a karaktert határozza meg.


  

	str_int(StringParam,IntParam)
	(string,integer):(i,o)(o,i)(i,i)

	(i,o)
	A StrParam által meghatározott decimális egész érték belsõ (bináris) értékét rendeli az IntParam-hoz.

	(o,i)
	IntParam decimális jegyeinek értékei rendeli a StringParam értékéhez.

	(i,i)
	Teljesül, ha a StrParam által meghatározott decimális egész érték belsõ (bináris) értéke megegyezik az IntParam által meghatározott számmal.


  

	str_len(String,Length)
	(string,integer):(i,i)(i,o)

	(i,i)
	Teljesül, ha a String-nek Length darab karaktere van.

	(i,o)
	Úgy teljesül, hogy Length-nek az String karaktereinek a számát adja.


  

	str_real(StringParam,RealParam)
	(string,real):(i,o)(o,i)(i,i)

	(i,o)
	A StrParam által meghatározott decimális valós érték belsõ (bináris) értékét rendeli a RealParam-hoz.

	(o,i)
	RealParam decimális jegyeinek értékei rendeli a StringParam értékéhez.

	(i,i)
	Teljesül, ha a StrParam által meghatározott decimális valós érték belsõ (bináris) értéke megegyezik a RealParam által meghatározott számmal.


  

	system(DosCommandString)
	(string):(i)

	(i)
	A DosCommandString parancsot küldi a Dos-nak végrehajtásra. 


  

	text 
	-

	(i)
	Visszaállítja a képernyõt szöveges üzemmódra. Nincs semmiféle hatása, ha már egyébként is ebben az üzemmódban van.


  

	time(Hours,Minutes,Seconds,Hundredths)
	(integer,integer,integer,integer):(i,i,i,i)(o,o,o,o)

	(i,i,i,i)
	Beállítja a számítógép belsõ óráján az idõt.

	(o,o,o,o)
	Kiolvassa a számítógép belsõ órájából az idõt.


  

	trace(Status)
	(symbol):(i)(o)

	(i)
	trace(on) bekapcsolja a nyomkövetést (abban az esetben, ha akár a trace, akár a shorttrace compiler parancs érvényben van), míg a trace(off) kikapcsolja a nyomkövetést.

	(o)
	Beállítja a Status-t on-ra vagy off-ra a nyomkövetés kapcsoltságának megfelelõen. 


  

	upper_lower(StringInUpperCase,StringInLowerCase)
	(string,string):(i,i)(i,o)(o,i)

	(i,i)
	Beállítja a második paraméter értékét az elsõ kisbetûs változatára.

	(i,o)
	Beállítja az elsõ paraméter értékét a második kisbetûs változatára.

	(o,i)
	Teljesül, ha az elsõ string a másodiknak a nagybetûs változata.


  

	window_attr(Attr)
	(integer):(i)

	(i)
	Az aktuális ablakhoz rendeli az Attr attribútum értéket.


  

	window_str(ScreenString)
	(string):(i)(o)

	(i)
	ScreenString értéke az aktuális ablak pillanatnyi tartalma lesz. Igy a ScreenString változónak ugyanannyi sora lesz, mint az ablaknak. Minden sor hossza az utolsó nem üres karakterig terjed.

	(o)
	ScreenString tartalma az ablakba íródik, az alábbi kritériumok teljesülése mellett:  

· Ha több sor van a stringben, mint amennyi sor az ablakban, akkor addíg íródnak a sorok, ameddíg az ablak engedi. 

· Ha kevesebb sor van a stringben, mint amennyi sor az ablakban, akkor az ablak maradék sorai szóközzel lesznek feltöltve. 

· Ha a string egy sorában több karakter van, mint az ablakban, akkor a sor kiesõ részét levágja. 

· Ha a string sorában kevesebb karakter van, mint az ablakban, akkor a sor további helyei szóközzel lesznek feltöltve. 


  

	write(e1,e2,e3,...,eN)
	( (i)... )

	(i)
	Kiírja a konstansokat vagy változókat az éppen aktív ablakba az aktuális kimeneti egységen. Tetszõleges számú argumentummal lehet hívni, amelyek mindegyike vagy konstans, vagy érvényes értékkel bíró, beépített alaptípusokból álló változó lehet.


  

	writedevice(SymbolicFileName)
	(symbol):(i)(o)

	(i)
	Átirányítja az aktuális kimeneti egységet a SymbolicFileName nevû megnyitott file-ba, amely vagy egy beépített szimbolikus file (screen vagy printer), vagy bármely a felhasználó által írásra vagy módosításra megnyitott file szimbolikus neve.

	(o)
	A SymbolicFileName az aktuális kimeneti egység szimbolikus neve lesz, amely vagy az elõre definiált screen vagy printer, vagy egy file.


  

	writef(FormatString,Arg1,Arg2,Arg3,...,ArgN)
	(i(,i))...)

	(i)
	Formázott kiviteli utasítás. Arg1-tõl ArgN-ig standard típusokból álló konstansok vagy változók állnak. A formátumstring tartalmaz változatlanul kinyomtatandó szokásos karaktereket és formátummeghatározó jeleket az alábbi módon:  

· - balra igazítást jelent (a jobbraigazítás a default helyzet)  

· m meghatározza a minimális mezõszélességet. 

· p megszabja a lebegõpontos számok ábrázolási pontosságát, vagy a stringbõl kiírandó karakterek maximális számát. Ennek a mezõnek tartalmaznia kell az f, e vagy g betûk egyikét, melyek jelentése: 
f A valósak fixpontos decimális alakban (default)  
e A valósak hatványkitevõs alakban  
g A legrövidebb alak használata

